MAPLE

Ce logiciel, développé au Canada, permet de faire du calcul symbolique et du calcul numérique. Il dispose en outre d'une excellente interface graphique.

Le minimum indispensable à connaître:

- ✓ démarrer Maple (mise en route d'un ordinateur et du logiciel sous Windows, ouverture et sauvegarde d'un fichier de travail)
- ✓ utiliser Maple pour calculer (l'invite - PROMPT - « > » qui indique que Maple attend vos instructions , la fin d'instruction « **;** ↵» (ou « **:** ↵» si l'on souhaite que la réponse soit cachée), le respect de la syntaxe)
- ✓ ne pas hésiter à utiliser l'aide (?, help, F1 après position du curseur sur le mot)
- ✓ utiliser les possibilités graphiques
- ✓ compléter le système avec l'apport de "librairies"(exemple en calcul matriciel : with linalg)
- ✓ savoir écrire ses propres fonctions et procédures
- ✓ connaître les grands principes algorithmiques (structure alternative, structure répétitive) pour programmer

Ce document vous aidera à prendre en main Maple: vous pouvez taper les exemples, noter en marge tout ce que vous rencontrez (affichage, résultats surprenants ...).
Il reprend une partie des TD que j'ai effectué en classes prépas au Lycée Daudet de Nîmes dans les années 1995.

Volontairement l'éventail mathématique proposé est large et peut même déborder vos préoccupations immédiates (arithmétique, géométrie...). Il est cependant instructif de piocher dans les sujets de Bac S ou ES et de vite se rendre compte de l'obsolescence de tels sujets avec l'invasion des calculatrices formelles et la mise à disposition de logiciels de calculs formels (Xcas par exemple est téléchargeable gratuitement).
A vos crayons afin d'imaginer de nouveaux sujets profitant de la puissance de tels outils.
Des exercices, indiqués avec le symbole ⊗ sont insérés ; certains sont abordables sans difficultés en seconde, certains ont été posé aux concours de CPGE PT(oral), PSI(oral ENSAM) ou parfois centrale, mines, polytechnique.

Cachez les solutions et laissez certains en attente pour y réfléchir plus tard – des notions utiles de programmation ne seront abordées que plus loin.

I) Quelques calculs pour débuter

\> 2+3;
↲ l’entrée déclenche:

- l’évaluation de l’expression tapée
- des modifications (simplification si possible)
- l’affichage du résultat,
- et à nouveau proposition de saisie avec l’invite >

\> ‘’ ; attention, c’est la dernière évaluation chronologique
\> 2+3 :
\> 1+2 ; 3*5 ; 2**3;

\> 2^3^4; normal, pas d’associativité !

Maple traduit l’expression tapée sous forme [opérateur, opérande].

\> 7 + 3*x ;



\> whattype (‘’) ;

donne le type général

op (expression) donne la liste des opérandes,
nops (expression) donne le nombre d’opérandes

\> op(7+3*x) ; op(2, 7+3*x) ; subs (x=cos(t), 7+3*x) ;

\> convert (3+x, ‘ * ‘);

Attention, pour la chaine ‘*’ ce sont les ‘ placés sur le clavier en dessous du 7

Les simplifications (une des grosses difficultés en calcul formel) ne sont pas toujours immédiates:

```
> sqrt(10^n + 2) - sqrt(10^n +1)  -  1/(sqrt(10^n+2) + sqrt(10^n+1));
  simplify(") ;
  exp(a+b) + exp(b);  expand("); factor(");
  sqrt(a+b) * sqrt(a-b);  combine(") ;
  A:= sqrt(x^2); assume(x>0): simplify(A);
```

$$\sqrt{10^n+2}-\sqrt{10^n+1}-\frac{1}{\sqrt{10^n+2}+\sqrt{10^n+1}}$$

$$0$$

$$e^{(a+b)}+e^b$$

$$e^a\,e^b+e^b$$

$$e^b\,(e^a+1)$$

$$\sqrt{a+b}\,\sqrt{a-b}$$

$$\sqrt{a^2-b^2}$$

$$A:=\sqrt{x^2}$$

$$x\sim$$

Voici les principaux simplificateurs:

Classe d'expressions	Fonction
entiers	simplification automatique
rationnels	simplification automatique
flottants	`evalf`
booléens	`evalb`
résidus mod p	`mod`
nombres p-adiques	`padic[evalp]`
matrices	`evalm`
fractions rationnelles	`normal`
développements limités	`series`
nombres algébriques	`evala`
racines carrées	`rationalize`
nombres complexes	`evalc`
fonction f	`simplify(...,f)`

```
> 1/(sqrt(2)+7);
```

$$\frac{1}{\sqrt{2}+7}$$

```
> rationalize(");
```

$$\frac{7}{47}-\frac{1}{47}\sqrt{2}$$

II) Les nombres entiers (type integer)

> restart :	très utile, réinitialise toutes les variables
> n := 100!;	100! est calculé et « stocké » dans la variable n attention à ne pas confondre « = » et « := »
> ifactor(n);	
> isolve(x*x - y*y = 9);	résolution de l'équation dans $\mathbb{Z}$
> binomial (25,10);	
> ? binomial	pour en savoir plus !
> bimonial(25,10)	observez Maple en cas d' erreur de syntaxe
> (2 + 3 * 5 ;	attention à la position de l 'erreur détectée!

D'autres essais

> type (10.3, integer) ;
> type (6,odd) ;
> is (6,even) ;
> type (17,prime) ;
> convert(256, binary) ;
> igcd(36,48) ;

⊗ Rechercher les triangles rectangles à côtés entiers d'hypoténuse 100.

```
> sol:={isolve (a**2 + b**2 = 10000)};
```

$sol := \{\{a=-100, b=0\}, \{a=100, b=0\}, \{b=-60, a=-80\}, \{b=80, a=-60\}, \{b=-80, a=60\}, \{a=-80, b=60\}, \{b=-28, a=-96\}, \{b=80, a=60\}, \{b=60, a=80\}, \{a=96, b=-28\}, \{b=100, a=0\}, \{a=-60, b=-80\}, \{b=-100, a=0\}, \{b=-60, a=80\}, \{b=96, a=-28\}, \{b=96, a=28\}, \{b=28, a=96\}, \{b=28, a=-96\}, \{a=-28, b=-96\}, \{a=28, b=-96\}\}$

```
> # on va trier ...
> for i from 1 to nops(sol)
  do
  if rhs(sol[i][1])>=0 and rhs(sol[i][2])>=0 then print(sol[i])
  fi
  od;
```

$$\{a=100, b=0\}$$
$$\{b=80, a=60\}$$
$$\{b=60, a=80\}$$
$$\{b=100, a=0\}$$
$$\{b=96, a=28\}$$
$$\{b=28, a=96\}$$

```
> |
```

⊗ Y a-t-il des rectangles à côtés entiers de diagonale $\sqrt{2009}$.

```
> an:=2009:
  for a from 1 to floor(sqrt(an))
  do
  b:= sqrt(an-a*a):
  if round(b)-b=0 then print(a,b) fi
  od:
```

```
28, 35
35, 28
```

Et de diagonale $\sqrt{2010}$? Et des pavés de diagonale $\sqrt{2009}$?
Y a-t-il eu jusqu'ici une année record en nombre de solutions ?

⊗ Quel est le plus petit nombre uniquement composé de 8 divisible par 2008 ?

```
a:=8:  an:=2008:
> for i  from 1 while (a mod an) <> 0 do a:=a*10+8 od:
> print(a,i);
```

```
88888888888888888888888888888888888888888888888888, 50
```

⊗ (oral 99-14) Etant donné un entier naturel a, on appelle diviseur propre de a tout diviseur différent de a.
Deux entiers naturels différents de zéro sont dits amiables si chacun d'eux est égal à la somme des diviseurs propres de l'autre.
Ecrire un programme qui détermine tous les couples d'entiers amiables inférieurs ou égaux à 1500.

⊗ Le polynôme d'Euler (1772) $n^2 + n + 41$ donne des nombres premiers pour n=0..40

```
> for i from 0 to 41 do i,isprime(i*i+i+41) od;
```

Rechercher les « nombres chanceux » (c'est-à-dire p tels que $n^2 + n + p$ …)
2,3,5,11,17,41 …

Une course à ce genre de polynôme existe aussi (45 consécutifs, Ruby 1989):

```
> i:=0:for n from -33 to 11  do k:=abs(36*n^2 + 18*n - 1801) :
  if  isprime(k)  then i:=i+1:print(n,i,k) fi; od:
```

```
-33, 1, 36809
-32, 2, 34487
-31, 3, 32237
-30, 4, 30059
-29, 5, 27953
-28, 6, 25919
```

⊗ Le problème de Fermat vient d'être résolu! Un autre "beau" problème se voit offert plusieurs prix pour sa résolution: c'est le problème de Collatz (d'après le professeur de Hambourg qui l'a lancé il y a quelques dizaines d'années). La suite est définie par récurrence u(n+1) = 3*u(n) + 1 si u(n) impair; u(n)/2 si u(n) pair.
La conjecture à montrer est: pour toute suite de Collatz, il existe k tel que u(k) = 1.
Vous pouvez tenter de balayer un intervalle, faire afficher les records (plus grand nombre atteint, plus grande suite ...) …

⊗ Cette conjecture de Fermat « les entiers $2^{(2^n)}+1$ sont premiers » s'est avérée fausse (à ce jour on ne sait si F(33) est premier ou pas).

```
> with(numtheory):
> for i from 1 to 6
  do i;k:=F(i);
  ifactor(k)
  od;
```

$$1$$
$$k := 5$$
$$(5)$$
$$2$$
$$k := 17$$
$$(17)$$
$$3$$
$$k := 257$$
$$(257)$$
$$4$$
$$k := 65537$$
$$(65537)$$
$$5$$
$$k := 4294967297$$
$$(641)\,(6700417)$$
$$6$$
$$k := 18446744073709551617$$
$$(67280421310721)\,(274177)$$

```
>
```

⊗ (oral 98-14) Ecrire une fonction f qui associe à l'entier n la somme des cubes des chiffres utilisés dans l'écriture de n en base 10.
Ecrire un programme permettant de déterminer tous les entiers n ≤ 1000
tels que f(n) = n.

```
· f:=proc (n)
  local z,s,k:
  z:=n:
  s:=0:
  k:=0;
  while (z>0)
  do    k:=irem (z,10, 'z');  s:= s + k**3;  od;
  RETURN (s);
  end:
· f(12);
```

9

```
· for j from 0 to 1000
  do
   if f(j) = j
     then print (j):
   fi:
  od;
```

0
1
153
370
371
407

⊗ Montrer que pour tout entier n, $n(n^6 - 1)$ est divisible par 7

```
> u:=n-> n*(n**6 -1);
```

$$u := n \rightarrow n\,(n^6 - 1)$$

```
> map(x->u(x) mod(7),[$1..30]);          # quelques essais
```

$$[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0]$$

```
> factor(u(n)) ;
```

$$n\,(n-1)\,(n+1)\,(n^2+n+1)\,(n^2-n+1)$$

```
> Factor(u(n)) mod(7) : sort(");
```

$$(n+2)\,(n+1)\,(n+4)\,(n+5)\,(n+6)\,(n+3)\,n$$

```
>    # ce produit de 7 entiers consécutif est divisible par 7
> msolve(u(n),7);
```

$$\{n = n\}$$

```
>    # tous les entiers sont solutions de cette équation modulo 7
```

⊗ (d’après oral centrale) f(n) est la somme des chiffres d’un entier n naturel écrit en base 10. Calculer $fofof\ (2011^{2011})$

```
> 
> f:=proc(n)
  local x,s:
  x:=n:s:=0:
  while x>0 do s:=s+irem(x,10): x:= iquo(x,10): od: RETURN(s) end:
> f (2011^2011);
                                                  29830
> f(");f(");
                                                  22
                                                  4
```

III) L’environnement Maple sous Windows

Vous avez à votre disposition une barre de menu

FILE: lecture, écriture sur disque; impression; essayez NEW et EXIT

EDIT: gestions de blocs (suppression, duplication);
le "copier coller" de Windows fonctionne.
EXECUTE est utile au chargement d'un fichier.
VIEW: personnalisation de la feuille de calcul (essayez Zoom Factor)
WINDOW: essayez TILE

HELP: l'aide directe sur le mot concerné par le curseur.
L’accès est possible avec CTRL-F1.
Observez les exemples proposés :
vous pouvez les ramener dans votre feuille de travail.
Avec "Topic Search" une recherche alphabétique est possible.
Essayez aussi Contents et son arborescence.

IV) Utilisation d’une librairie

Maple possède des milliers de fonctions, mais elles ne sont pas toutes chargées automatiquement en mémoire.

On a rencontrera par exemple :
with (plots) pour compléter les fonctions graphiques,
with (linalg) pour compléter l’ algèbre linéaire…

```
> with(numtheory):
> divisors(9876543210123456789);
```

{84414899231824417, 9876543210123456789, 740740741, 39, 1, 3, 9, 13, 9411851848793, 3292181070041152263, 253244697695473251, 759734093086419753, 1097393690013717421, 743301, 82589, 367062222102927, 28235555546379, 84706666639137, 122354074034309, 1101186666308781, 247767, 19059, 57177, 57777777759, 4444444443, 13333333329, 19259259253, 173333333277, 56980057, 8969, 172736296240157, 1554626666161413, 13287407403089, 1049373, 2222222223, 518208888720471, 39862222209267, 119586666627801, 26907, 80721, 116597, 6666666669, 170940171, 512820513, 349791, 117, 1481481481, 6353}

```
with (combinat):powerset({a,b,c,d});
```

$\{\{\,\}, \{a,b,c,d\}, \{b,c,d\}, \{c,d\}, \{a,c,d\}, \{d\}, \{a,d\}, \{b,d\}, \{a,b,d\}, \{a,b,c\}, \{b,c\}, \{c\}, \{a,c\}, \{a\}, \{b\}, \{a,b\}\}$

Géométrie

restart : with(geometry) :
triangle (T, [point (A, -1,-1), point (B,1,4), point (C,7,0)]) :
midpoint (M1,A,B) : midpoint (M2,C,B) : midpoint (M3,A,C) :
orthocenter (H,T) :
incircle(inc,T,’centername’ = O) :
excircle (obj,T,[ex1(I1) , ex2(I2) , ex3(I3)]) ;
EulerCircle(ec,T,’centername’= K) ;
AreTangent(ec,inc) ; AreTangent(ec,ex1) ;
draw ([inc,ec(color=blue,filled=true),ex1,ex2,ex3,T]) ;

C’est le théorème de Feuerbach : le cercle d’Euler d’un triangle est tangent aux cercles inscrit et exinscrits.

Sa propre bibliothèque !

Quand on travaille dans un domaine particulier, il peut être intéressant de regrouper un ensemble de procédures personnelles.

```
> restart:ex_bidon := table () ;
ex_bidon := table([
  ])
> ex_bidon [bonjour] := proc()  print (`Comment allez vous !`) end :

> bonjour();
                                                    bonjour( )
> with (ex_bidon);
                                                    [bonjour]
> bonjour();
                                                 Comment allez vous !
```

Qu'il est possible de sauvegarder (libname, save…)

V) Divers types que l'on peut manipuler avec Maple

Les rationnels et les réels (float), Variables et affectation

> mon_calcul := 1/2+2/3;
affectation de la partie droite à la variable identifiée ' mon_calcul ' ;
une simplification de base est faite ;
mon_calcul pointe alors sur 7/6 (mon_calcul → 7/6)

> whattype(");
> evalf(mon_calcul); force l'évaluation dans les réels
> whattype(");
> Sum(k/(k+1) , k =1..20) = sum(k/(k+1) , k =1..20);
> evalf(rhs("));
> whattype(sqrt(2));
> alpha := x; MAJ-ENTREE pour changer de ligne
x := 3 ;
alpha ; (alpha → x → 3)

> X ; x ; Maple différencie majuscule et minuscule
> x := 'x' ; x ; pour « réinitialiser la variable x »

```
> sqrt(7)=cfrac(sqrt(7),8);
```

$$\sqrt{7} = 2 + \cfrac{1}{1 + \cfrac{1}{1 + \cfrac{1}{1 + \cfrac{1}{4 + \cfrac{1}{1 + \cfrac{1}{1 + \cfrac{1}{1 + \cfrac{1}{4 + \dots}}}}}}}}$$

```
>
```

Attention aux mots réservés

```
> pi; evalf(");
> Pi; evalf(");
> evalf(Pi,1000);
> Digits := 20; evalf(Pi);
> x:= 1/ (sqrt(2) -1);    evalf(x,20);
> exp (1);
> sin(3*Pi/5);
```

```
> evalf(exp(Pi*sqrt(163)),28);
```

$$.2625374126407687440000000000\ 10^{18}$$

bigre! avec 28 décimales ?!

Les nombres algébriques   (racines de polynômes)

```
alias(alpha=RootOf(x^3+2*x^2+1,x)):
59*alpha^4/(alpha^5+alpha^3+1);
```

$$59 \frac{\alpha^4}{\alpha^5 + \alpha^3 + 1}$$

```
evala(");
```

$$-6\,\alpha - 9\,\alpha^2 + 8$$

Les complexes   (complex)

> z := (1 + I) ^2 / (1 - 2 * I);
> argument(z); conjugate(z) ; Re(z); Im (z) ; abs(z);

⊗ (oral 98-1) Déterminer les nombres complexes tels que z²/(2z+3i) soit imaginaire pur. Les représenter dans le plan complexe.

```
z:=a+I*b:
C:= z**2/(2*z+3*I);
f:=unapply (simplify(evalc(Re(C))),a,b);
solve (f(a,b) =0);                    # on enlèvera z=0!
with(plots):implicitplot(f(a,b)=0,a=-15..15,b=-15..15,grid=[100,100],axes=BOXED);
```

$$C := \frac{(a+Ib)^2}{2a+2Ib+3I}$$

$$f := (a, b) \to 2\frac{a(a^2+b^2+3b)}{4a^2+4b^2+12b+9}$$

$$\{a=0, b=b\}, \{a=\sqrt{-b^2-3b}, b=b\}, \{a=-\sqrt{-b^2-3b}, b=b\}$$



Visiblement une droite a = 0 et la courbe a²+b²+3b=0

⊗ (oral 99-1) Dans $\mathbb{C}$ résoudre l'équation $z^2+\overline{z}+iz=0$. Montrer que les points dont les affixes sont solutions forment un triangle rectangle.

```
z:=a+I*b:
C:= z**2 + a-I*b + I*z ;
r:=evalc(Re(C)); i:=evalc(Im(C));
solve ({r=0,i=0});
```

$$C := (a+Ib)^2 + a - Ib + I(a+Ib)$$
$$r := a^2 - b^2 + a - b$$
$$i := 2\,a\,b - b + a$$
$$\{a=0, b=0\}, \{a=-1-\mathrm{RootOf}(2\_Z^2+4\_Z+1), b=\mathrm{RootOf}(2\_Z^2+4\_Z+1)\}$$

```
> with(geometry):
Sol:= {solve(2*x*x+4*x+1)};
point(A,0,0):
point(B,-1-op(1,Sol),op(2,Sol)):
point(D,-1-op(2,Sol),op(1,Sol)):
triangle(T,[A,B,D]):IsRightTriangle(T);
```

$$Sol := \{-1+\frac{1}{2}\sqrt{2}, -1-\frac{1}{2}\sqrt{2}\}$$
$$true$$

Variables indicées

> x[a]; a:=3; x[a+1];

Tableaux

t := array(1..3, 1..2 , [[4,5] , [0,4] , [7,-1]]) ;
print (t) ; t[2,1] ;

Séquences (suite = exprseq)

x := a,b,c ;
whattype(x) ;
x[1] ;
seq (j^3, j=1..4) ;

Ensembles

x := {alpha,bof, 1,bof} ;
whattype(x) ;   ne pas se fier à l'ordre !
nops(x) ; op(x) ;   on récupère la suite
y := x union {26} ;

Listes

suite ordonnée d'expression
L := [a,b,c,d] ; whattype(L) ; nops(L) ;L[3] ; op(L) ; L2 := [op(L),e] ;

VI) L'analyse

Fonctions à une variable et graphisme

> restart
> h(x) := x+7 ;   pas conseillé !
> g := x -> (x^3 - 2*x - 1) / (x^2 + x + 1);   mieux
> g(5);

Limit (f(x) , x=infinity)= limit (f(x) , x=infinity);

f := x -> 1/x ; readlib(singular) : singular(f(x),x);
readlib(iscont) : iscont(f(x),x=0..2);
iscont(f(x),x=0..2, closed);   à manier avec réserves !
D(f);   l   a dérivée   f'
diff (f(x) , x) ; D(f) (x) ;   f' (x) de deux manières

series (tan(sin(x))-sin(tan(x)) , x=0 , 8);   développement limité
Int (g(x) , x = -1..1) = int (g(x) , x = -1..1);
Int(g(x) , x)= int(g(x) , x);
plot(g, -5..5);   ou bien   plot (g(x) , x=-5..5)
des possibilités existent concernant les axes, les repères...

plot ({sin, cos} , 0..2*Pi, -1..1);
plot (tan(x), x=-3..3, y=-4..4, discont = true);

k := unapply(2*x-7,x) ;   fonction à partir d'une expression

sin (Pi/ 3) ;   comment est-ce possible ! !
op(4, eval(sin)) ;

restart ; k := x-> sin(x)/x ; limit (k(x) , x=0) ;
k(0) := 1 ; op(4, eval(k)) ; k(0) ;

plot ({1/x, sin(x)} , x=0..1) ;   tracé multiple

utile aussi
f :=x -> if x>0 then sin(x) else 0 fi;
plot(f);

Simplification des fonctions:

Commande	Résultat
`expand(···)`	$\sin(a+b) \mapsto \sin(a)\cos(b)+\cos(a)\sin(b)$
	$\cos(a+b) \mapsto \cos(a)\cos(b)-\sin(a)\sin(b)$
	idem pour les fonctions hyperboliques
	$e^{a+b} \mapsto e^a e^b$
`combine(···,trig)`	$\cos(a)\cos(b) \mapsto \cos(a-b)/2+\cos(a+b)/2$
	$\cos(a)\sin(b) \mapsto \sin(a+b)/2-\sin(a-b)/2$
	$\sin(a)\sin(b) \mapsto \cos(a-b)/2-\cos(a+b)/2$
	idem pour les fonctions hyperboliques
`combine(···,exp)`	$e^a e^b \mapsto e^{a+b}$
	$(e^a)^b \mapsto e^{ab}$
	$e^{a+n\ln b} \mapsto b^n e^a$ où n est entier
`combine(···,ln)`	$n\ln b \mapsto \ln(b^n)$ où n est entier
	$\ln a+\ln b \mapsto \ln(ab)$
`simplify(···,trig)`	$\sin(x)^2+\cos(x)^2 \mapsto 1$
	$\cosh(x)^2-\sinh(x)^2 \mapsto 1$
`simplify(···,exp)`	$e^{a\ln b} \mapsto b^a$
`simplify(···,ln)`	$\ln(b^a) \mapsto a\ln b$
ou `expand(···)`	$\ln(ab) \mapsto \ln a+\ln b$
`simplify(···,power)`	$x^a x^b \mapsto x^{a+b}$
`simplify(···,power,symbolic)`	$(a^b)^c \mapsto a^{bc}$
ou `combine(···,power)`	$\sqrt{a^2} \mapsto a$
`convert(···,exp)`	$\cos(x) \mapsto (e^{ix}+e^{-ix})/2$
	$\cosh(x) \mapsto (e^x+e^{-x})/2$
`convert(···,trig)`	$e^{ix} \mapsto \cos(x)+i\sin(x)$
	$e^x \mapsto \cosh(x)+\sinh(x)$
`convert(···,ln)`	$\arccos(x) \mapsto -i\ln(x+i\sqrt{1-x^2})$
	$\operatorname{arctanh}(x) \mapsto (\ln(1+x)-\ln(1-x))/2$

⊗ Quel est le volume du plus gros colis que l'on puisse envoyer en chronopost sachant que les dimensions doivent respecter

$$a + 2b + 2c \leq 300$$

```
> L:=300-2*H-2*K:
> volume:=L*H*K:
> maximize(volume);
```

250000

⊗ (oral 99-10)

Soit f la fonction: $(a,b) \to \int_0^{+\infty} (x^2 + ax + b)^2 e^{-2x} dx$

En quel(s) point(s) la fonction f atteint-elle son minimum? Le calculer.

```
> restart:f:=(a,b)->int((x*x+a*x+b)**2*exp(-2*x), x=0..infinity);
```

$$f := (a,b) \to \int_0^{\infty} (x^2 + a\,x + b)^2\, \mathrm{e}^{(-2\,x)}\, dx$$

```
> plot3d(f(a,b),a=-2..2,b=-2.
```



```
> minimize(f(a,b));
```

$$\frac{1}{8}$$

```
> da:=D[1](f):db:=D[2](f):s:=solve({da(a,b)=0,db(a,b)=0},{a,b}):   #CN mais non CS
> assign(s);
> f(a,b);
```

$$\frac{1}{8}$$

⊗ (oral 98-8) Déterminer les coefficients a,b,c,d et e pour que la fonction f définie par

$$f(x) = \cos x - \frac{a + bx^2 + cx^4}{1 + dx^2 + ex^4}$$

soit un infiniment petit d'ordre le plus élevé possible au voisinage de zéro.
Donner alors un équivalent de f.

⊗ Calculer la limite de ((tan(sin(x))-sin(tan(x))) / x^7 en 0. Expliciter en donnant les développement de Taylor de x→tan(sin(x)) et x→sin(tan(x)) en 0 à un ordre suffisant.

⊗ Donner le développement de Taylor de x→sin(x) en 0. Convertir en polynôme puis tracer en même temps les deux courbes.

⊗ (oral 99-12)

Trouver tous les polynômes P dans $\mathbb{R}[X]$ tels que la série de terme général $u_n = (n^7 + 3n^6)^{1/7} - (P(n))^{1/3}$ soit convergente

le polynôme est nécessairement de degré < 4

```
> restart;
  P(x) := a*x^3 + b*x^2 + c* x +d:
  g:=x-> (x^7 + 3* x^6) ^(1/7):
> f:= unapply(g(x) - ( P(x)) ^(1/3), x);
```

$$f := x \to (x^7+3x^6)^{1/7} - (a x^3 + b x^2 + c x + d)^{1/3}$$

```
> series(f(x),x=infinity,4);
```

$$(1-a^{1/3})x - \frac{1}{3}\frac{b}{a^{2/3}} + \frac{3}{7} + \frac{-a^{1/3}\left(\frac{1}{3}\frac{c}{a} - \frac{1}{9}\frac{b^2}{a^2}\right) - \frac{27}{49}}{x} + \frac{-a^{1/3}\left(\frac{1}{3}\frac{d}{a} - \frac{2}{9}\frac{b\,c}{a^2} + \frac{5}{81}\frac{b^3}{a^3}\right) + \frac{351}{343}}{x^2} + \frac{-\frac{5265}{2401} - a^{1/3}\left(-\frac{2}{9}\frac{b\,d}{a^2} - \frac{1}{9}\frac{c^2}{a^2} + \frac{5}{27}\frac{b^2 c}{a^3} - \frac{10}{243}\frac{b^4}{a^4}\right)}{x^3} + O\left(\frac{1}{x^4}\right)$$

donc a=1

```
> a:=1:
> f(x):
> series(f(x),x=infinity,5);
```

$$-\frac{1}{3}b + \frac{3}{7} + \frac{-\frac{1}{3}c + \frac{1}{9}b^2 - \frac{27}{49}}{x} + \frac{-\frac{1}{3}d + \frac{2}{9}b\,c - \frac{5}{81}b^3 + \frac{351}{343}}{x^2} + \frac{\frac{2}{9}b\,d + \frac{1}{9}c^2 - \frac{5}{27}b^2 c + \frac{10}{243}b^4 - \frac{5265}{2401}}{x^3} + \frac{\frac{2}{9}c\,d - \frac{5}{81}b^2 d - \frac{10}{81}b\,c^2 + \frac{5}{9}\left(-\frac{2}{9}b\,d - \frac{1}{9}c^2\right)b + \frac{40}{243}b^3 c - \frac{22}{729}b^5 + \frac{85293}{16807}}{x^4} + O\left(\frac{1}{x^5}\right)$$

```
> b:=9/7:
  series(f(x),x=infinity,5);
  c:=-54/49:
  series(f(x),x=infinity,5);
  limit(f(x),x=infinity);
```

$$\frac{-\frac{18}{49} - \frac{1}{3}c}{x} + \frac{-\frac{1}{3}d + \frac{2}{7}c + \frac{306}{343}}{x^2} + \frac{-\frac{4995}{2401} + \frac{2}{7}d + \frac{1}{9}c^2 - \frac{15}{49}c}{x^3} + \frac{\frac{83511}{16807} + \frac{2}{9}c\,d - \frac{15}{49}d - \frac{5}{21}c^2 + \frac{120}{343}c}{x^4} + O\left(\frac{1}{x^5}\right)$$

$$\frac{-\frac{1}{3}d + \frac{198}{343}}{x^2} + \frac{-\frac{3861}{2401} + \frac{2}{7}d}{x^3} + \frac{\frac{72171}{16807} - \frac{27}{49}d}{x^4} + O\left(\frac{1}{x^5}\right)$$

$$0$$

⊗ (oral 98-9) Pour quelles valeurs de a, b et c les primitives de la fonction f définie par $f(x) = \dfrac{ax^2+bx+c}{(x-1)^3(x+2)^5}$ sont-elles des fractions rationnelles?
Déterminer une de ces primitives.

Courbe polaire

with(plots) : n'est pas dans la bibliothèque standard
polarplot (sin(3*t), t=0..2*Pi) ;



Fonctions à plusieurs variables

restart; f := (x,y) -> 8*x*x – y*y + 2*x ;
diff(f(x,y) , x) ;
diff (g(x,y) , x$2, y$4) ;
plot3d(f, -2..2,-3..3) ;

```
> restart:with(plots):
  rad:=proc(aa) aa/360*2*Pi end:
  x:=D*(A*sin(beta)*cos(Theta)+R*cos(s+phi)*cos(Theta+Omega)
    -R*sin(mu)*sin(s+phi)*sin(Theta+Omega))*exp(Theta*cot(alpha)):
  y:=(-A*sin(beta)*sin(Theta)-R*cos(s+phi)*sin(Theta+Omega)
    -R*sin(mu)*sin(s+phi)*cos(Theta+Omega))*exp(Theta*cot(alpha)):
  z:=(-A*cos(beta)+R*sin(s+phi)*cos(mu))*exp(Theta*cot(alpha)):
  SR:=R=RE+k:
  SRE:=RE=1/sqrt(cos(s)^2/a^2+sin(s)^2/b^2):
  Sk:=k=L*exp(-(2*(s-P)/W[1])^2)*exp(-(2*g/W[2])^2):
  Sg:=g=2*Pi/N*(Theta*N/2/Pi-round(Theta*N/2/Pi)):S:={D=1,alpha=rad(83),beta=rad(90),phi=rad(1),mu=rad(1),
  Omega=rad(1),smM=rad(-170)..rad(170),A=2.5,a=1,b=0.9,P=10,W[1]=100,W[2]=20,N=15,L=1/2}:
  R1:=simplify(subs(S,subs(SR,SRE,Sk,Sg,[x,y,z]))):
  plot3d(R1,s=subs(S,smM),Theta=-1*Pi..4*Pi,grid=[40,60],style=patchnogrid,scaling=constrained,lightmodel=light
  2,shading=xyz,projection=0.5,orientation=[-97,58]);
```



Courbes paramétrées

plot ([sin**3 , cos*(1- cos) , 0..2*Pi]);



⊗ (oral 99-7) Soit la courbe de représentation paramétrique
$x = u^3/(u^2-9)$ $y=u(u-2)/(u-3)$. Représenter cette courbe ; préciser les asymptotes, les points doubles et les points d'inflexion.

⊗ (oral 98-13) Tracer la courbe paramétrée définie par :
$x = (2t-1)/(t^2-1)$ $y =(t^2+3)/(t-1)$
Etudier les branches infinies. Calculer les coordonnées du point double.

Famille de fonctions

```
p:= t -> t * exp(x) + x^2;
plot ({p(-1) , p(1),  p(2)} , x=-3..3 , y= -5..5);

k := int(x^2 * sin(x-a) ,x) ;
value(‘’) ;  k := simplify(‘’) ;
with (plots) : animate(k, x=-Pi..Pi, a=0..1) ;
```

Fonction implicite

```
with(plots) :   f:= (x,y)->x^2 + 2*y^2-1;
implicitplot (f, -2..2, -2..2, scaling=constrained);
```

Superposition de graphiques

```
f1 := plot (1/x, x=-3..3, y= -4..4) ;
f2 := implicitplot(x^3 + y^4 – x , x=-3..3, y=-3..3) ;
display({f1,f2}) ;                                 on a d’abord préparé les tracés
```

Résolution d' équations

```
solve (a*x^2 + b*x + c, x);   solve (x^3 -1);
```

Trigonométrie

```
sin (x ^2) + cos (x ^2); simplify (");
sin (x) ^2 + cos (x) ^2;   simplify (");
expand(cos(3*x)); expand(tan(3*x));
combine (cos(a)*cos(b) – sin(a)*sin(b) , trig) ;

Int(1/(1+cos(a)*cos(x)) , x=0..Pi) = int(1/(1+cos(a)*cos(x)) , x=0..Pi);
assume (a, real) :                                         il faut aider maple!
int(1/(1+cos(a)*cos(x)) , x=0..Pi);  simplify(‘’) ;
```

Polynômes, fractions rationnelles

```
restart: (x+y)^10:
expand(");
factor(");
P:= x^8-1:factor(P);
factor(P,I);
factor(P,sqrt(2));
factor(P,{I,sqrt(2)});
```

$$x^{10}+10\,x^9\,y+45\,x^8\,y^2+120\,x^7\,y^3+210\,x^6\,y^4+252\,x^5\,y^5+210\,x^4\,y^6+120\,x^3\,y^7+45\,x^2\,y^8+10\,x\,y^9+y^{10}$$

$$(x+y)^{10}$$

$$(x-1)\,(x+1)\,(x^2+1)\,(x^4+1)$$

$$(x^2+I)\,(-x^2+I)\,(-x+I)\,(x+I)\,(x+1)\,(x-1)$$

$$-(x^2+1)\,(-x^2+\sqrt{2}\,x-1)\,(x^2+\sqrt{2}\,x+1)\,(x+1)\,(x-1)$$

$$-\frac{1}{16}(-x+I)\,(x+I)\,(2\,x-\sqrt{2}+I\sqrt{2})\,(-2\,x+\sqrt{2}+I\sqrt{2})\,(-2\,x-\sqrt{2}+I\sqrt{2})\,(2\,x+\sqrt{2}+I\sqrt{2})\,(x+1)\,(x-1)$$

.

```
> restart:
  readlib(cost):
  P:=x^4 - 2*x^3 + x^2 - 2*x + 1;
  cost(P);
  factor(P,sqrt(2));
```

$$P := x^4-2\,x^3+x^2-2\,x+1$$

$$4\,additions+8\,multiplications$$

$$(x^2-x+\sqrt{2}\,x+1)\,(x^2-x-\sqrt{2}\,x+1)$$

```
> convert(P,horner);
  cost(");
```

$$1+(-2+(1+(-2+x)\,x)\,x)\,x$$

$$4\,additions+3\,multiplications$$

```
>  P:= x*2; x:=1; P; x:=2; P;
> restart; P : = (x+1)/((x+3)*(x+2)^2);
> expand(P);numer(P); denom(P); convert(P,parfrac,x);

> Q := (x^4 - y^4) / (x^2 - y^2); normal (Q);
> x := 5 : y := 3 : Q;
>normal ((x^4-y^4)/(x^2-y^2)) ;
```

Réécritures de polynômes :

Polynôme p	$zx^2+x^2-(x^2+y^2)(ax-2by)+zy^2+y^2$
collect(p,x)	$(z+1+2by)x^2-y^2ax+2y^3b+zy^2+y^2-x^3a$
collect(p,[x,y])	$(z+1+2by)x^2-y^2ax+2y^3b+(z+1)y^2-x^3a$
collect(p,[x,y], distributed)	$(z+1)x^2+2byx^2-y^2ax+2y^3b+(z+1)y^2-x^3a$
expand(p)	$zx^2+x^2-x^3a+2x^2by-y^2ax+2y^3b+zy^2+y^2$
factor(p)	$(x^2+y^2)(-ax+z+1+2by)$

Réécritures de fractions rationnelles :

Fraction f	Opération	Résultat
$\frac{x^3+3x^2+2x+yx^2+3yx+2y}{x^3+yx+2x^2+2y}$	`normal(f)`	$\frac{x^2+yx+x+y}{x^2+y}$
	`factor(f)`	$\frac{(x+1)(x+y)}{x^2+y}$
$\frac{x^3+3x^2+2x+yx^2+3yx+2y}{x^3+x+2x^2+2}$	`collect(f,y)`	$\frac{(x^2+3x+2)\,y}{x^3+x+2x^2+2}+\frac{x^3+3x^2+2x}{x^3+x+2x^2+2}$
	`collect(f,y,normal)`	$\frac{(x+1)y}{x^2+1}+\frac{(x+1)x}{x^2+1}$
$\frac{x^2+3yx+2}{x^2+1}$	`expand(f)`	$\frac{x^2}{x^2+1}+\frac{3yx}{x^2+1}+\frac{2}{x^2+1}$
$\frac{3x^2y+3xy^2+1}{x^2-4y^2}$	`convert(f,parfrac,x)`	$3y+\frac{1+18y^3}{4y(x-2y)}-\frac{1+6y^3}{4y(x+2y)}$

⊗ Interpolation de Lagrange (polynôme passant par des points distincts) ; on retrouve les coefficients des polynômes L dans le problème de l'inversion d'une matrice de Vandermonde.

```
> restart:f:=x->10+1/(1+x*x):
> X:=[1,2,3,4,5]:n:=nops("):
> Y:=map(f,X):
> P:=x->sum(a[j]*x^j, j=0..n-1):
> eq:=seq(P(X[i])=Y[i], i=1..n):
> solve({eq});
```

$$\{a_4=\frac{19}{4420}, a_3=\frac{-147}{2210}, a_0=\frac{2485}{221}, a_1=\frac{-2373}{2210}, a_2=\frac{1731}{4420}\}$$

```
> assign(");
> P(x);
```

$$\frac{2485}{221}-\frac{2373}{2210}x+\frac{1731}{4420}x^2-\frac{147}{2210}x^3+\frac{19}{4420}x^4$$

```
> interp(X,Y,x);
```

$$\frac{2485}{221}-\frac{2373}{2210}x+\frac{1731}{4420}x^2-\frac{147}{2210}x^3+\frac{19}{4420}x^4$$

```
> plot({f,P},-1..9,color=BLACK);
```




```
> for i from 1 to n do
  L[i](x):=product((x-X[j])/(X[i]-X[j]),j=1..i-1)*product((x-X[j])/(X[i]-X[j]),j=i+1..n);expand(");od;
```

```
> for i from 1 to n do
  L[i](x):=product((x-X[j])/(X[i]-X[j]),j=1..i-1)*product((x-X[j])/(X[i]-X[j]),j=i+1..n);expand(");od;
>
```

$$L_1(x) := (-x+2)\left(-\frac{1}{2}x+\frac{3}{2}\right)\left(-\frac{1}{3}x+\frac{4}{3}\right)\left(-\frac{1}{4}x+\frac{5}{4}\right)$$

$$\frac{1}{24}x^4 - \frac{7}{12}x^3 + \frac{71}{24}x^2 - \frac{77}{12}x + 5$$

$$L_2(x) := (x-1)(-x+3)\left(-\frac{1}{2}x+2\right)\left(-\frac{1}{3}x+\frac{5}{3}\right)$$

$$-\frac{1}{6}x^4 + \frac{13}{6}x^3 - \frac{59}{6}x^2 + \frac{107}{6}x - 10$$

```
> Q:=x->sum(Y[k]*L[k](x),k=1..n):simplify(Q(x));
```

$$\frac{19}{4420}x^4 - \frac{147}{2210}x^3 + \frac{1731}{4420}x^2 - \frac{2373}{2210}x + \frac{2485}{221}$$

```
> with(linalg):V:=vandermonde(X);
```

$$V := \begin{bmatrix} 1 & 1 & 1 & 1 & 1 \\ 1 & 2 & 4 & 8 & 16 \\ 1 & 3 & 9 & 27 & 81 \\ 1 & 4 & 16 & 64 & 256 \\ 1 & 5 & 25 & 125 & 625 \end{bmatrix}$$

```
> inverse(V);
```

$$\begin{bmatrix} 5 & -10 & 10 & -5 & 1 \\ \frac{-77}{12} & \frac{107}{6} & \frac{-39}{2} & \frac{61}{6} & \frac{-25}{12} \\ \frac{71}{24} & \frac{-59}{6} & \frac{49}{4} & \frac{-41}{6} & \frac{35}{24} \\ \frac{-7}{12} & \frac{13}{6} & -3 & \frac{11}{6} & \frac{-5}{12} \\ \frac{1}{24} & \frac{-1}{6} & \frac{1}{4} & \frac{-1}{6} & \frac{1}{24} \end{bmatrix}$$

⊗ Comment doit-on choisir m pour que le polynôme $X^3 + X^2 + mX + 6$ admette deux zéros a et b tels que a+b=ab ?

```
>
> solve({a+b+c=-1,a*b+b*c+c*a=m,a*b*c=-6,a+b=a*b});
```

$\{b = \mathrm{RootOf}(\_Z^2 - 2\_Z + 2), a = 2 - \mathrm{RootOf}(\_Z^2 - 2\_Z + 2), m = -4, c = -3\}, \{b = \mathrm{RootOf}(\_Z^2 + 3\_Z - 3), m = -9, a = -\mathrm{RootOf}(\_Z^2 + 3\_Z - 3) - 3, c = 2\}$

```
> solve(X^3+X^2-4*X+6);
```

$$-3, 1+I, 1-I$$

```
> solve(X^3+X^2-9*X+6);
```

$$2, -\frac{3}{2}+\frac{1}{2}\sqrt{21}, -\frac{3}{2}-\frac{1}{2}\sqrt{21}$$

```
>
```

⊗ (oral 99-2) On considère le polynôme réel suivant $X^4 + X^3 + aX^2 + \sqrt{2}X + b$
Déterminer a et b pour que 1+i soit zéro de P ; calculer alors tous les zéros de P
Factoriser P en facteurs irréductibles dans $\mathbb{R}[X]$ et dans $\mathbb{C}[X]$.

```
restart:
P:= x^4 + x^3 + a*x^2 + sqrt(2)*x + b;
Q:=subs(x=1+I,P);
simplify(evalc(Re(Q)));
b:=solve (");

simplify(evalc(Im(Q)));
a:=solve (");
```

$$P := x^4 + x^3 + a\,x^2 + \sqrt{2}\,x + b$$
$$Q := -6 + 2\,I + 2\,I\,a + (1+I)\sqrt{2} + b$$
$$-6 + \sqrt{2} + b$$
$$b := 6 - \sqrt{2}$$
$$2 + \sqrt{2} + 2\,a$$
$$a := -1 - \frac{1}{2}\sqrt{2}$$

```
P;
```

$$x^4 + x^3 + \left(-1 - \frac{1}{2}\sqrt{2}\right)x^2 + \sqrt{2}\,x + 6 - \sqrt{2}$$

```
factor(P);
```

$$\frac{1}{2}(x^2 - 2\,x + 2)\,(2\,x^2 + 6\,x + 6 - \sqrt{2})$$

```
factor(P,I);
```

$$\frac{1}{4}(x - 1 - I)\,(x - 1 + I)\,(2\,x + 3 + I - I\sqrt{2})\,(2\,x + 3 - I + I\sqrt{2})$$

⊗ forme explicite de la suite de Fibonacci donnée sous forme de récurrence linéaire

$$\begin{cases} u_0 = 0, \quad u_0 = 1 \\ u_n = u_{n-1} + u_{n-2}\,, \; pour\; n \geq 2 \end{cases}$$

```
> f:=rsolve({u(n)=u(n-1)+u(n-2),u(0)=0,u(1)=1},u); g:=unapply(f,n): g(5)
```

$$f := \frac{\left(-1 + \frac{1}{5}\sqrt{5}\right)\left(-\frac{2}{-\sqrt{5}+1}\right)^n}{-\sqrt{5}+1} + \frac{\left(-1 - \frac{1}{5}\sqrt{5}\right)\left(-\frac{2}{\sqrt{5}+1}\right)^n}{\sqrt{5}+1}$$

$$-32\,\frac{-1 + \frac{1}{5}\sqrt{5}}{(-\sqrt{5}+1)^6} - 32\,\frac{-1 - \frac{1}{5}\sqrt{5}}{(\sqrt{5}+1)^6}$$

```
> simplify(");
```

$$\frac{20480}{(\sqrt{5}-1)^6\,(\sqrt{5}+1)^6}$$

```
> normal(",expanded);
```

$$5$$

```
>
```

ou pour changer, la suite de Padovan, ou la suite de Perrin…

Le nombre d’or et le nombre d’argent :

```
> solve (x*x=x+1,x);evalf(");
```

$$\frac{1}{2}\sqrt{5}+\frac{1}{2}, \frac{1}{2}-\frac{1}{2}\sqrt{5}$$

$$1.618033989, -.6180339890$$

```
> [solve (x*x*x=x*x+x+1,x)];
```

$$\left[\frac{1}{3}\%2+\frac{4}{3}\%1+\frac{1}{3}, -\frac{1}{6}\%2-\frac{2}{3}\%1+\frac{1}{3}+\frac{1}{2}I\sqrt{3}\left(\frac{1}{3}\%2-\frac{4}{3}\%1\right), -\frac{1}{6}\%2-\frac{2}{3}\%1+\frac{1}{3}-\frac{1}{2}I\sqrt{3}\left(\frac{1}{3}\%2-\frac{4}{3}\%1\right)\right]$$

$$\%1 := \frac{1}{(19+3\sqrt{33})^{1/3}}$$

$$\%2 := (19+3\sqrt{33})^{1/3}$$

```
> op(1,");
```

$$\frac{1}{3}(19+3\sqrt{33})^{1/3}+\frac{4}{3}\frac{1}{(19+3\sqrt{33})^{1/3}}+\frac{1}{3}$$

```
> evalf(");
```

$$1.839286755$$

résolution numérique :

```
> f2:=x->x*x-x-1:
  f3:=x->x*x*x-x*x-x-1:
  f4:=x->x*x*x*x-x*x*x-x*x-x-1:
  f5:=x->x*x*x*x*x-x*x*x*x-x*x*x-x*x-x-1:

  plot({f2,f3,f4,f5},0..2,-7..2);
```




```
> fsolve(f5(x),x,0..2);
```

$$1.965948237$$

⊗ Deux échelles mesurant 2 et 3m sont placées dans un couloir aux murs verticaux et au sol horizontal.



EF = 1 m. AF=y, FB=z. Quelle est la largeur x du couloir ?

```
>
> fsolve ({x=y*sqrt(4.0-x*x),x=z*sqrt(9.0-x*x),x=y+z},{x,y,z},{x=1..5});
```

$$\{x = 1.231185724, z = .4500403039, y = .7811454199\}$$

⊗ Une chèvre est attachée au bord d'un champ circulaire. Quelle doit être la longueur de la corde pour que la chèvre ne broute que la moitié du champ ?



La chèvre!

Rayon du champ de centre O: 1

longeur de la corde attachée au point A sur la circonférence : L

à l'extrémité de sa corde la chèvre coupe le cercle en B et C et le diamètre (OA) en D

OAB est isocèle les angles OAB et AOB mesurent alpha et beta

```
> restart;
  fsolve ({2*alpha+beta=Pi,sin(beta) = L*sin(alpha), beta+L*L*alpha - sin(beta) = Pi/2},{alpha,beta,L},{beta=0.5..1.5});
```

$$\{L = 1.158728473, \alpha = .9528478648, \beta = 1.235896924\}$$

VII) Les équations différentielles

Maple sait résoudre certaines équations différentielles avec **DSOLVE** Un classique **Y' = Y**

```
> restart:
  dsolve (diff(y(x),x) - y(x) , y(x));
```

$$y(x) = e^{x}\_C1$$

```
> subs(_C1=alpha,rhs(")):
  sol:=unapply(",alpha,x):
> plot(sol(3,x),x=-3..3,y=0..30);
```




```
> plot({seq(sol(k,x), k=-2..3)},x=-3..3,y=-10..10, color=black);
```



Autre exemple avec **condition initiale**

```
> restart: dsolve ( { diff(y(x),x) - 2*y(x) = 1 , y(0) = 7} , y(x));
```

$$y(x) = -\frac{1}{2} + \frac{15}{2}e^{(2x)}$$

ou encore

```
> dsolve ( { diff(y(x),x) - 2*y(x) = 1 , D(y)(-2)=4} , y(x)):
  simplify(");
```

$$y(x) = -\frac{1}{2} + 2e^{(2x+4)}$$

Dans le cas où on MAPLE ne peut trouver de solution on peut préciser des **options de résolution**

```
> eq:= D(y)(x) - x**2 * y(x) = 1;
> dsolve ( eq , y(x)):
```

$$eq := D(y)(x) - x^2\,y(x) = 1$$

BOF!

```
> Order:= 12:
  dsolve ( eq , y(x),series);
```

$$y(x) = y(0) + x + \frac{1}{3}y(0)\,x^3 + \frac{1}{4}x^4 + \frac{1}{18}y(0)\,x^6 + \frac{1}{28}x^7 + \frac{1}{162}y(0)\,x^9 + \frac{1}{280}x^{10} + O(x^{12})$$

```
> sol:=dsolve ( { eq , y(1)=2},y(x),numeric):
> sol(2);
```

$$[x = 2, y(x) = 25.65270163099149]$$

```
> plots [odeplot] (sol,[x,y(x)] , -3 .. 1);
>
```



Autre exemple: **une intégrale**

```
> restart:
  Int (1/ (t**6 + 1) , t=0..x):
  f:= unapply(int (1/ (t**6 + 1) , t=0..x),x):
> eq := diff(y(t) , t) = 1/ (t**6 + 1);
```

$$eq := \frac{\partial}{\partial t} y(t) = \frac{1}{t^6 + 1}$$

```
> sol_exa:= dsolve ({eq, y(0) = 0}, y(t)):
  plot_exa:= plot (f,-10..10, style = point, symbol=circle,color=black):
```

résolution numérique Runge - Kutta rkf45

```
> sol_num:=dsolve ({eq, y(0) = 0}, y(t) , type=numeric):
  plot_num:= plots [odeplot] (sol_num,[t,y(t)],-10..10,color=red):
  plots[display]({plot_exa,plot_num});
```



PAS MAL!

Méthode numériques (solutions approchées)

METHODE DE PICARD

équation différentielle y' = f(x,t)

```
> restart:
  f:= (x,t) -> 2*x*t:
> eq := diff(y(t) , t) = f (t , y(t) );
```

$$eq := \frac{\partial}{\partial t} y(t) = 2\,t\,y(t)$$

condition initiale x=0 , y=1

```
> dsolve ({eq, y(0) = 1}, y(t));
```

$$y(t) = e^{(t^2)}$$

```
> sol := rhs("):
```

On construit une suite telle que p0(x) = Y0   p1(x) = Y0 + int(f(u, p0(u)) u= X0..x)   p2(x) = Y0 + int(f(u, p1(u)) u=X0..x) ....

```
> PICARD := proc(g , X0 , Y0 , t, ordre)
  local u,i,s:
  s:= Y0:
  for i to  ordre
  do
     s:= subs(t=u,s):
     s:= Y0 + int (g(u,s) , u=X0..t):
  od:
  s
  end:
> p:= PICARD (f,0,1,t,3);
```

$$p := 1 + \frac{1}{6} t^6 + \frac{1}{2} t^4 + t^2$$

```
> plot ({p , sol}, t = -1.5..1.5,y = 0..8, color = black);
```



autre exemple $Y' = \sin(t)\ Y + t$

```
> F:= (x,t) -> sin(x) * t + x :
> eq := diff(y(t) , t) = F (t,y(t) ):
```

```
> p:= PICARD (F,0,0,t,3):
> plot (p, t=-3..3,y=0..10,color=black):
>
```

METHODE D'EULER

```
>
> restart:
  with(plottools):
  f:= (x,t) -> 2*t/x:
> eq := diff(y(t) , t) = f (t , y(t) );
```

$$eq := \frac{\partial}{\partial t}\mathrm{y}(t) = 2\,\frac{\mathrm{y}(t)}{t}$$

condition initiale x=0.1 , y=0.01

```
> dsolve ({eq, y(0.1) = 0.01}, y(t)):
> sol := rhs("):
```

On construit les points (X0,Y0) (X0+h, Y0 + f(X0)h) , (X0+2h, Y1 + f(X1)h) ....

```
> S:={}:
  EULER := proc(g,X0,Y0,pas,couleur)
  global S:
  local i,X,Y:
  X:=X0: Y:=Y0:
  for i to  3/pas
  do
    Y:= Y +  pas * g(X,Y):
    X := X + pas:
    S:= S union {point([X,Y],color=couleur)}:
  od:
  end:
> EULER (f,0.1,0.01,0.1, green):
  courbe:=plot (sol, t = 0..3,y = 0..9,color=black):
> EULER (f,0.1,0.01,0.05, blue):
  EULER (f,0.1,0.01,0.01, red):
> plots[display](courbe,S);
```




```
>
```

METHODE D'EULER - INFLUENCE DU PAS

```
> f:= (x,t) -> t**2-x:
```

RICCATI

```
> eq := diff(y(t) , t) = f (t , y(t) ):
> l:={}:
  EULER := proc(g,X0,Y0,pas,couleur,style)
  global l:
  local i,X,Y,X2,Y2:
  X:=X0: Y:=Y0:
  for i while (Y<40 and Y > -40 and  X<4.5)
  do
    Y2:= Y +  pas *g(X,Y):
    X2 := X + pas:
    l:= l union {line([X,Y],[X2,Y2], color=couleur, linestyle=style)}:
    X:=X2:Y:=Y2
  od:
  end:
```

condition initaile 0.7 0.5

```
EULER (f,0.7,0.5,1.4, black,2):
EULER (f,0.7,0.5,1.2, blue,3):
> EULER (f,0.7,0.5,0.15, red,1):
  plots[display](l);
```



```
> f:= (x,t) -> -t+1/x:
```

EQUATION D'EULER

```
> eq := diff(y(t) , t) = f (t , y(t) ):
```

condition initaile 0.7 0.5

```
> EULER (f,0.7,0.2,4, black,2):
  EULER (f,0.7,0.2,3, blue,3):EULER (f,0.7,0.2,1, red,1):
  plots[display](l);
```



ACCELERATION

```
>
> f:= (x,t) -> 2*t/x:
> eq := diff(y(t) , t) = f (t , y(t) ):
```

condition initiale x=0.1 , y=0.01

```
> dsolve ({eq, y(0.1) = 0.01}, y(t)):
> sol := rhs("):
```

On accélère (méthode des trapèzes) y (n+1) := y(n) + int (f(x,y) de x(n) à x(n+1)) évaluée en y(n) + h/2 (f(x(n),y(n) + f(x(n+1),y(n+1))
en estimant grossièrement y(n+1) par y(n) + h f (x(n) , y(n))

```
> S:={}:
  EULER := proc(g,X0,Y0,pas,couleur)
  global S:
  local i,X,Y,a,b:
  X:=X0: Y:=Y0:
  for i to 3/pas
  do
    a:= X+pas:
    b:= Y + pas *g(X,Y):
    Y:= Y + pas/2 * (g(X,Y) + g(a,b)):
    X := a:
    S:= S union {point([X,Y],color=couleur)}:
  od:
  end:
> EULER (f,0.1,0.01,0.1, green):
  courbe:=plot (sol, t = 0..3,y = 0..9,color=black):
> EULER (f,0.1,0.01,0.05, red):
> plots[display](courbe,S);
```



AUTRE ACCELERATION

>

On accélère autrement (méthode de la tangente) y (n+1) évalué en y(n) + h f(x(n)+ h/2,y(x(n)+h/2))

en estimant grossièrement y(x(n) + h/2) par y(n) + h/2 y'(n)) donc finalement y (n+1) := y(n) + h f (x(n)+ h/2,y(n)+h/2 f(x(n),y(n))

```
> S:={}:
  EULER := proc(g,X0,Y0,pas,couleur)
  global S:
  local i,X,Y,a,b:
  X:=X0: Y:=Y0:
  for i to 3/pas
  do
    a:= X+pas/2:
    b:= Y + pas/2 *g(X,Y):
    X := X + pas:
    Y:= Y + pas* g(a,b):
    S:= S union {point([X,Y],color=couleur)}:
  od:
  end:
> EULER (f,0.1,0.01,0.1, green):
  courbe:=plot (sol, t = 0..3,y = 0..9,color=black):
> EULER (f,0.1,0.01,0.05, red):
> plots[display](courbe,S):
```

⊗ (oral 98-10) Résoudre l'équation différentielle suivante:

$$xy'(x)+(1-x)y(x)=\frac{xe^x}{x^2+1}$$

Tracer quelques courbes intégrales.

Existe-t-il des solutions continues sur 3 ? Justifier la réponse.

⊗ (oral 98-12) Résoudre l'équation différentielle suivante:

$$y''(x) + x\sin(y(x)) = 0$$

avec les conditions initiales y(0) = 0 et y'(0) = 0,5
Soit f la solution. Tracer son graphe.
Déterminer une valeur approchée à 10^{-3} près de la plus petite valeur strictement positive annulant f.

⊗ (oral 98-11)
La valeur pour x=1 de la solution de l'équation différentielle $y''(x)+(y'(x))^2=1$
avec les conditions initiales y(0) = 0 et y'(0) = a
est une fonction f du paramètre réel a.
Tracer le graphe de f pour a variant de -1 à 10.

⊗ (oral 99-11)
1) Résoudre l'équation différentielle $xy' + \frac{3x+4}{2(x+1)}y = \frac{x}{\sqrt{x+1}}$

2) Existe-t-il des solutions sur]-1,+∞[?
Si oui, tracer le graphe de telles solutions.

⊗ exemple d'influence aux conditions initiales : résoudre de façon numérique
y'(x) – 3y(x) + 3x=0 tester avec des conditions initiales y(0) = 0.333, 1/3, 0.334

⊗ Problème de poursuite

Désignons par $x(t)$, $y(t)$ et v_0 les coordonnées cartésiennes et la vitesse du jardinier, puis par $X(t)$, $Y(t)$ et V_0 les coordonnées cartésiennes et la vitesse du chien. Nous pouvons écrire :

$$\left[\frac{dX}{dt}\right]^2 + \left[\frac{dY}{dt}\right]^2 = V_0^2$$

d'où :

$$\frac{dX}{dt} = \pm\sqrt{V_0^2 - \left[\frac{dY}{dt}\right]^2} \quad \text{et} \quad \frac{dY}{dt} = \pm\sqrt{V_0^2 - \left[\frac{dX}{dt}\right]^2},$$

ainsi que :

$$\left[\frac{dx}{dt}\right]^2 + \left[\frac{dy}{dt}\right]^2 = v_0^2.$$

Par ailleurs, nous savons que le chien se dirige en permanence vers le jardinier, nous pouvons alors écrire :

$$\frac{dY}{dX} = \frac{Y-y}{X-x} \quad \text{d'où} \quad \frac{dY}{dt} = \frac{Y-y}{X-x}\frac{dX}{dt}.$$

On écrit l'équation horaire du jardinier :

$$x = a\cos[f(t)] \quad \text{et} \quad y = b\sin[f(t)]$$

$$\frac{dx}{dt} = -af'(t)\sin[f(t)] \quad \text{et} \quad \frac{dy}{dt} = bf'(t)\cos[f(t)]$$

$$\left[\frac{dx}{dt}\right]^2 + \left[\frac{dy}{dt}\right]^2 = v_0^2 = f'^2(t)[a^2 + b^2].$$

De là on tire que $f'(t)$ est une constante et par conséquent que $f(t) = \omega t + \varphi$, ce qui donne :

$$\omega = \sqrt{\frac{v_0^2}{a^2 + b^2}}$$

et $\varphi = 0$ (condition initiale). Donc, à $t = 0$: $y = -b\cos(\omega t)$ et $x = a\sin(\omega t)$.

Il faudra prendre garde au signe de dY/dt ainsi que celui de X dans l'exécution des calculs et l'on devra prévoir des tests pour affecter les bons signes.

VII) Programmer

Créer une procédure

```
> bonjour := proc()
print (`coucou`)
end :
```

que l'on pourra utiliser

`> bonjour() ;` provoquera l'affichage : coucou

Créer une fonction

```
k :=(x,y) -> x+y ;
k(5,4) ;

f := x-> ln(x) ;
g := x-> 1/exp(x) ;
(f@g)(x) ;                          pas terrible !
simplify('') ;                      on est dans les complexes !
simplify(assume(x,real),'') ;
```

avec une procédure

```
> f := proc(x)
local z ;
z := x*2 + 3 ;
RETURN (1/z)
end ;
```

z est une variable locale, interne à la procédure

x est un paramètre utilisé à l'appel de la fonction

`> f(7) ;` provoquera l'affichage 1/17

Lecture d'une procédure Maple

eval(arcsin); interface(verboseproc=2); eval(arcsin);

Structure alternative

```
> maximo := proc (x,y)
local z ;
if x > y
   then z := x
   else z := y
fi
end ;
```

la condition x > y   est vrai ou fausse : c'est un booléen
autres connecteurs logiques utilisables : or, and, not

```
> Pi = 1 ;
> evalb('') ;
```

Structure répétitive à compteur

```
> for i from 0 to 5 by 2
  do
     print (i^3);
  od;

> s:=0:
for k from 1 to 100
  do s:= s+k
od:
's' = s;
'k' = k;

> for i from 1 to 4
  do
    expand( cos(i*x));
    subs( cos(x)=t, '');
  od;
```

dans tous ces cas on connaît à l'avance
le nombre de passages dans la boucle

```
moyenne := proc()
  local m, n;
  m:=0;
  if nargs = 0 then ERROR(`Calcul impossible: pas de valeurs!`)
  else
    for n to nargs
      do m := m + args[n]
      od;
    RETURN(m/nargs);
  fi;
end:
```

```
> moyenne();
> moyenne(1,2,5,100);
> trace(moyenne); moyenne(1,2,5,100); untrace(moyenne);
```

Structure répétitive à test

```
s:=0;
for k  while k<=100
  do s:= s+k
  od:
's' = s;
```

le test est un booléen

```
s:=0;
for k while s<1997
  do s:= s+k
  od:
's' =s;
'k' = k;
```

Variables globales, locales

source d'erreurs !

```
> restart ;
 a :=1 ;                                   global
 p := proc()
 a := 7
 end :

> a ; p() ; a ;                            modifié !

> q := proc()
   local a
   a := 25
   end :
> a ;  q() ; a ;

> r := proc(x)
   x := x+1 ;
   end ;
>r(3) ;
```

provoque une erreur, car lors de l'appel de la procédure, le paramètre formel x est remplacé (une fois et définitivement) par 3.
Dans la ligne apparaît donc la ligne 3 := 3+1   incorrecte

Programmes récursifs

```
> fact := proc (x)
if x = 0 then RETURN (1)
else RETURN(x * fact(x-1))
fi
end :

> fact(5) ;
```

```
  ackerman := proc(x,y)
  if x=0  then y+1
  elif y=0 then ackerman(x-1,1)
  else ackerman(x-1, ackerman(x,y-1))
  fi:
  end:
>
> ackerman(3,2);
                                                        29
> printlevel:=1000:
> ackerman(1,2);
{--> enter ackerman, args = 1, 2
{--> enter ackerman, args = 1, 1
{--> enter ackerman, args = 1, 0
{--> enter ackerman, args = 0, 1
                                                        2
<-- exit ackerman (now in ackerman) = 2}
                                                        2
<-- exit ackerman (now in ackerman) = 2}
{--> enter ackerman, args = 0, 2
                                                        3
<-- exit ackerman (now in ackerman) = 3}
                                                        3
<-- exit ackerman (now in ackerman) = 3}
{--> enter ackerman, args = 0, 3
                                                        4
<-- exit ackerman (now in ackerman) = 4}
                                                        4
<-- exit ackerman (now at top level) = 4}
                                                        4
```

⊗ Ecrire une procédure qui résout une équation du 1° degré

⊗ Ecrire une procédure qui échange deux variables a et b ;   que fait ceci ?
a :=a+b ; b := a-b ; a :=a-b ;

⊗ (oral 98-15) Ecrire une fonction qui associe à toute suite finie de nombres réels , l'indice de la plus grande valeur. Si plusieurs valeurs sont égales à cette valeur maximale, le résultat est l'indice le plus petit.

VI) Algèbre linéaire

⊗ L'application $f(x,y,z) = (x + 2y - 3z, x - y + z)$ de $\mathbb{R}^3$ $dans$ $\mathbb{R}^2$ est-elle linéaire ; donner noyau et image.

```
> restart:a:=array(1..3):b:=array(1..3):
> f:=V->([V[1]+2*V[2]-3*V[3], V[1]-V[2]+V[3]]);
```

$$f := V \to [V_1 + 2V_2 - 3V_3, V_1 - V_2 + V_3]$$

```
> x:=f(evalm(alpha*a+beta*b));
```

$$x := [\alpha a_1 + \beta b_1 + 2\alpha a_2 + 2\beta b_2 - 3\alpha a_3 - 3\beta b_3, \alpha a_1 + \beta b_1 - \alpha a_2 - \beta b_2 + \alpha a_3 + \beta b_3]$$

```
> y:=evalm( (alpha *f(a))  +  beta *f(b));
```

$$y := [\alpha(a_1 + 2a_2 - 3a_3) + \beta(b_1 + 2b_2 - 3b_3), \alpha(a_1 - a_2 + a_3) + \beta(b_1 - b_2 + b_3)]$$

```
>
> map(expand,evalm(x-y));
```

$$[0, 0]$$

```
> with(linalg):M:=transpose(matrix([f([1,0,0]),f([0,1,0]),f([0,0,1])]));
Warning, new definition for norm
Warning, new definition for trace
```

$$M := \begin{bmatrix} 1 & 2 & -3 \\ 1 & -1 & 1 \end{bmatrix}$$

```
> kernel(M);colspace(M);
```

$$\left\{\left[\frac{1}{3}, \frac{4}{3}, 1\right]\right\}$$

$$\{[0, 1], [1, 0]\}$$

```
>
```

Calcul matriciel

with (linalg) : A := matrix ([[1,2,3] ,[4,5,6]]) ; A[2,2] ;

c'est un tableau dont l'indice commence à 1

B := matrix (3,3,0) ;

C := matrix (4,4,(i,j)->3*i-j) ;

det (C) ;

V :=vector (4,[1,2,0,1]) ; V[3] ;

norm(U,1) ;   somme des valeurs absolues

norm(U,2) ;   racine de la somme des carrés

norm(U,infinity) ; max des valeurs absolues ;

C &* V ; evalm ('') ;

Lors de la résolution de système linéaire il est fortement déconseillé d'utiliser $X = A^{-1}B$. Le calcul de l'inverse n'étant pas toujours valable.

Il vaut mieux s'inspirer de la méthode du pivot de Gauss.

```
> restart:with(linalg):
  A:=matrix([[1,2,2,2],[1,3,-2,-1],[3,5,8,8]]);
  for i from 1 to 3 do
  L[i] := sum(A[i,j]*X[j],j=1..3)=A[i,4]: print(L[i]);
  od:
  solve({L[1],L[2],L[3]});
Warning, new definition for norm
Warning, new definition for trace
```

$$A := \begin{bmatrix} 1 & 2 & 2 & 2 \\ 1 & 3 & -2 & -1 \\ 3 & 5 & 8 & 8 \end{bmatrix}$$

$$X_1 + 2\,X_2 + 2\,X_3 = 2$$

$$X_1 + 3\,X_2 - 2\,X_3 = -1$$

$$3\,X_1 + 5\,X_2 + 8\,X_3 = 8$$

$$\{X_1 = 3, X_2 = -1, X_3 = \frac{1}{2}\}$$

```
> for j from 1 to 4 do A[2,j]:=-A[1,j]+A[2,j];od:
  for j from 1 to 4 do A[3,j]:=3*A[1,j]-A[3,j];od:print(A);
```

$$\begin{bmatrix} 1 & 2 & 2 & 2 \\ 0 & 1 & -4 & -3 \\ 0 & 1 & -2 & -2 \end{bmatrix}$$

en pratique, on cherche d'abord le pivot de valeur absolue maxi (pour des raisons de précision), avec échange de lignes

```
> A:= pivot ( A, 2,2);
```

$$A := \begin{bmatrix} 1 & 0 & 10 & 8 \\ 0 & 1 & -4 & -3 \\ 0 & 0 & 2 & 1 \end{bmatrix}$$

```
> for i from 1 to 3 do
  print(sum(A[i,k]*X[k],k=1..3)=A[i,4]);
  od:
```

$$X_1 + 10\,X_3 = 8$$
$$X_2 - 4\,X_3 = -3$$
$$2\,X_3 = 1$$

Reprenons ces manipulations sous forme d'opérations matricielles

```
> restart:with(linalg):
  A:=matrix([[1,2,2,2],[1,3,-2,-1],[3,5,8,8]]);
  B1:=matrix([[1,0,0],[-1,1,0],[0,0,1]]);
  A:=evalm(B1 &*A);
Warning, new definition for norm
Warning, new definition for trace
```

$$A := \begin{bmatrix} 1 & 2 & 2 & 2 \\ 1 & 3 & -2 & -1 \\ 3 & 5 & 8 & 8 \end{bmatrix}$$

$$B1 := \begin{bmatrix} 1 & 0 & 0 \\ -1 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$A := \begin{bmatrix} 1 & 2 & 2 & 2 \\ 0 & 1 & -4 & -3 \\ 3 & 5 & 8 & 8 \end{bmatrix}$$

c'était la manipulation L2 := L2 - L1; maintenant L3 := L3 -3* L1

```
> B2:=matrix([[1,0,0],[0,1,0],[-3,0,1]]):
  A:=evalm(B2 &*A):
```

observons B2*B1

```
> evalm(B2&*B1):
```

astuce de la matrice bordante

```
> A:=concat(diag(1,1,1),matrix([[1,2,2,2],[1,3,-2,-1],[3,5,8,8]])):
  pivot(",1,4):
```

Bien sur Maple connait tout ça!

```
> restart:with(linalg):
  A:=matrix([[1,2,2,2],[1,3,-2,-1],[3,5,8,8]]):
```

```
> gausselim(A);
```

$$\begin{bmatrix} 1 & 2 & 2 & 2 \\ 0 & 1 & -4 & -3 \\ 0 & 0 & -2 & -1 \end{bmatrix}$$

```
> gaussjord(A);
```

$$\begin{bmatrix} 1 & 0 & 0 & 3 \\ 0 & 1 & 0 & -1 \\ 0 & 0 & 1 & \frac{1}{2} \end{bmatrix}$$

⊗ Tester le théorème sur les matrices de Vandermonde,pour n=5
« Si $z \in \mathbb{C}$ est une racine n-ième de l'unité, alors : z^{-1} est une racine n-ième de l'unité, et on trouve : $V(\alpha).V(\alpha^{-1}) = n\,I$. »

```
restart:n:=5:r:=exp(I*2*Pi/n);evalc(");
```

$$r := e^{\left(\frac{2}{5} I\pi\right)}$$

$$\frac{1}{4}\sqrt{5} - \frac{1}{4} + \frac{1}{4} I \sqrt{2}\sqrt{5+\sqrt{5}}$$

```
with(linalg):V1:=vandermonde([seq(r^k,k=0..n-1)]):
for i from 1 to n do for j from 1 to n do V1[i,j]:=rationalize(evalc(V1[i,j])) od od:
evalm(V1);
```

$$\begin{bmatrix} 1 & 1 & 1 & 1 & 1 \\ 1 & \%2 & \%4 & \%5 & \%6 \\ 1 & \%4 & \%6 & \%2 & \%5 \\ 1 & \%5 & \%2 & \%6 & \%4 \\ 1 & \%6 & \%5 & \%4 & \%2 \end{bmatrix}$$

$$\%1 := I\sqrt{2}\sqrt{5+\sqrt{5}}$$

$$\%2 := \frac{1}{4}\sqrt{5} - \frac{1}{4} + \frac{1}{4}\%1$$

$$\%3 := I\sqrt{5}\sqrt{2}\sqrt{5+\sqrt{5}}$$

$$\%4 := -\frac{1}{4} - \frac{1}{4}\sqrt{5} + \frac{1}{8}\%3 - \frac{1}{8}\%1$$

$$\%5 := -\frac{1}{4} - \frac{1}{4}\sqrt{5} + \frac{1}{8}\%1 - \frac{1}{8}\%3$$

$$\%6 := \frac{1}{4}\sqrt{5} - \frac{1}{4} - \frac{1}{4}\%1$$

```
> invr:=1/r:rationalize(evalc("));"^n;rationalize(evalc("));
```

$$\frac{1}{4}\sqrt{5} - \frac{1}{4} - \frac{1}{4} I\sqrt{2}\sqrt{5+\sqrt{5}}$$

$$\left(\frac{1}{4}\sqrt{5} - \frac{1}{4} - \frac{1}{4} I\sqrt{2}\sqrt{5+\sqrt{5}}\right)^5$$

$$1$$

```
> V2:=vandermonde([seq(invr^k,k=0..n-1)]):for i from 1 to n do for j from 1 to n do
  V2[i,j]:=rationalize(evalc(V2[i,j])) od od:
  evalm(V2):
> M:=evalm(V1&*V2):for i from 1 to n do for j from 1 to n do M[i,j]:=rationalize(evalc(M[i,j])) od od:
  evalm(M);
```

$$\begin{bmatrix} 5 & 0 & 0 & 0 & 0 \\ 0 & 5 & 0 & 0 & 0 \\ 0 & 0 & 5 & 0 & 0 \\ 0 & 0 & 0 & 5 & 0 \\ 0 & 0 & 0 & 0 & 5 \end{bmatrix}$$

⊗ Ecrire une fonction qui teste si une matrice est un « carré magique »
(somme des termes diagonaux égaux aux sommes des lignes et aux sommes des colonnes)

⊗ (oral Mines) valeurs propres et vecteurs propres de …

```
> restart: with(linalg):

  A:=matrix([[a*a,a*b,a*b,b*b],[a*b,a*a,b*b,a*b],[a*b,b*b,a*a,a*b],[b*b,a*b,a*b,a*a]]);
Warning, new definition for norm
Warning, new definition for trace
```

$$A := \begin{bmatrix} a^2 & a\,b & a\,b & b^2 \\ a\,b & a^2 & b^2 & a\,b \\ a\,b & b^2 & a^2 & a\,b \\ b^2 & a\,b & a\,b & a^2 \end{bmatrix}$$

```
> eigenvects(A);
```

$$[b^2 - 2\,a\,b + a^2, 1, \{[1, -1, -1, 1]\}], [a^2 - b^2, 2, \{[0, -1, 1, 0], [-1, 0, 0, 1]\}], [b^2 + 2\,a\,b + a^2, 1, \{[1, 1, 1, 1]\}]$$

```
> # reste à étudier les cas particuliers où le v.p. sont confondues
> solve(a*a-b*b=b*b-2*a*b+a*a);solve(a*a-b*b=b*b+2*a*b+a*a);solve(b*b-2*a*b+a*a=b*b+2*a*b+a*a);
```

$$\{b = 0, a = a\}, \{a = b, b = b\}$$

$$\{b = 0, a = a\}, \{b = b, a = -b\}$$

$$\{b = b, a = 0\}, \{b = 0, a = a\}$$

```
> b:=a:A:=matrix([[a*a,a*b,a*b,b*b],[a*b,a*a,b*b,a*b],[a*b,b*b,a*a,a*b],[b*b,a*b,a*b,a*a]]);eigenvects(A);
```

$$A := \begin{bmatrix} a^2 & a^2 & a^2 & a^2 \\ a^2 & a^2 & a^2 & a^2 \\ a^2 & a^2 & a^2 & a^2 \\ a^2 & a^2 & a^2 & a^2 \end{bmatrix}$$

$$[4\,a^2, 1, \{[1, 1, 1, 1]\}], [0, 3, \{[0, -1, 0, 1], [0, -1, 1, 0], [1, -1, 0, 0]\}]$$

```
> b:=-a:A:=matrix([[a*a,a*b,a*b,b*b],[a*b,a*a,b*b,a*b],[a*b,b*b,a*a,a*b],[b*b,a*b,a*b,a*a]]):eigenvects(A);
```

$$[4\,a^2, 1, \{[1, -1, -1, 1]\}], [0, 3, \{[1, 0, 1, 0], [-1, 0, 0, 1], [1, 1, 0, 0]\}]$$

```
> a:=0:b:=0:A:=matrix([[a*a,a*b,a*b,b*b],[a*b,a*a,b*b,a*b],[a*b,b*b,a*a,a*b],[b*b,a*b,a*b,a*a]]);
```

$$A := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

⊗ (oral Mines) CNS pour que cette matrice d'ordre n soit inversible et expression de K^{-1} lorsqu'elle existe (ici un exemple avec n=6 peut mettre sur la piste)

```
n:=7:
K:= matrix (n,n):        # rempli avec b
for i to n
do for j to n
  do if i<>j then K[i,j] := b else  K[i,j] := a fi
  od;
od;
```

Warning, new definition for norm
Warning, new definition for trace

```
> evalm(K);
```

$$\begin{bmatrix} a & b & b & b & b & b & b \\ b & a & b & b & b & b & b \\ b & b & a & b & b & b & b \\ b & b & b & a & b & b & b \\ b & b & b & b & a & b & b \\ b & b & b & b & b & a & b \\ b & b & b & b & b & b & a \end{bmatrix}$$

```
> det(K);
```

$$a^7 - 21\,b^2\,a^5 + 70\,b^3\,a^4 - 105\,b^4\,a^3 + 84\,b^5\,a^2 - 35\,b^6\,a + 6\,b^7$$

```
> factor(");
```

$$(a+6\,b)\,(a-b)^6$$

```
> inverse(K);
```

$$\begin{bmatrix} \frac{a+5\,b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} \\ -\frac{b}{\%1} & \frac{a+5\,b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} \\ -\frac{b}{\%1} & -\frac{b}{\%1} & \frac{a+5\,b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} \\ -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & \frac{a+5\,b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} \\ -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & \frac{a+5\,b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} \\ -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & \frac{a+5\,b}{\%1} & -\frac{b}{\%1} \\ -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & -\frac{b}{\%1} & \frac{a+5\,b}{\%1} \end{bmatrix}$$

$$\%1 := a^2 + 5\,b\,a - 6\,b^2$$

⊗ (oral Polytech.) Déterminer les éléments propres de cette matrice d'ordre n. (ici un exemple avec n=5 peut mettre sur la piste)

```
> restart: with(linalg):
  n:=5:
  K:= matrix (n,n):
  for i to n
  do for j to n
    do if abs(i-j)=1 then K[i,j] := a elif i=j then  K[i,j]:= b else  K[i,j] := 0 fi
    od;
  od;

Warning, new definition for norm
Warning, new definition for trace
> evalm(K);
```

$$\begin{bmatrix} b & a & 0 & 0 & 0 \\ a & b & a & 0 & 0 \\ 0 & a & b & a & 0 \\ 0 & 0 & a & b & a \\ 0 & 0 & 0 & a & b \end{bmatrix}$$

```
> charpoly(K,x);factor(");eigenvects(K);
```

$$x^5 - 5x^4 b + 10x^3 b^2 - 4x^3 a^2 - 10x^2 b^3 + 12x^2 b a^2 + 5x b^4 - 12x b^2 a^2 + 3x a^4 - b^5 + 4b^3 a^2 - 3b a^4$$

$$-(-x+b)(-b+a+x)(b+a-x)(-b^2+2xb+3a^2-x^2)$$

$$[b, 1, \{[1, 0, -1, 0, 1]\}], [a+b, 1, \{[-1, -1, 0, 1, 1]\}], [b+\sqrt{3}\,a, 1, \{[1, \sqrt{3}, 2, \sqrt{3}, 1]\}], [b-\sqrt{3}\,a, 1, \{[1, -\sqrt{3}, 2, -\sqrt{3}, 1]\}], [-a+b, 1, \{[1, -1, 0, 1, -1]\}]$$

⊗ Calcul de valeur propre (méthode de Souriau 1948)

Soit $\mathbf{A}$ une matrice carrée d'ordre n dont on se propose de calculer les valeurs propres λj, lesquelles sont les racines du polynôme caractéristique :

$$P_n(x) = a_0 x^n + a_1 x^{n-1} + a_2 x^{n-2} + \cdots + a_n = 0 \quad \text{avec} \quad a_0 = 1.$$

Nous nous proposons de calculer directement les coefficients a_k puis de calculer les racines du polynôme. Pour ce faire, considérons une suite de matrices B_q définie de la façon suivante :

$$\mathbf{B}_q = \mathbf{A}^q + a_1 \mathbf{A}^{q-1} + a_2 \mathbf{A}^{q-2} + \cdots + a_q \mathbf{I},$$

où $\mathbf{I}$ est la matrice unité d'ordre n.

a **–** Donner la relation de récurrence qui existe entre deux matrices consécutives B_k et B_{k+1}. Quelle valeur convient-il de donner à $\mathbf{B}_0$?

b **–** Démontrer les relations de Newton exprimant les coefficients a_k en fonction des sommes des racines élevées à la même puissance m. On notera :

$$S^m = \sum_{j=1}^{n} \lambda_{j^m} = \mathrm{Tr}(\mathbf{A}^m).$$

En utilisant l'expression de a_k donnée par la relation de Newton, montrer que

$$a_k = -\mathrm{Tr}(\mathbf{A}\mathbf{B}_{k-1})/k,$$

et déduire une procédure générale de calcul des valeurs propres d'une matrice.

a **–** Nous avons :

$$B_q = A^q + a_1 A^{q-1} + a_2 A^{q-2} + \cdots + a_q I$$
$$B_{q+1} = A^{q+1} + a_1 A^q + a_2 A^{q-1} + \cdots + a_{q+1} I = A(A^q + a_1 A^{q-1} + a_2 A^{q-2} + \cdots + a_q) + a_{q+1} I$$

d'où la relation de récurrence : $B_{q+1} = AB_q + a_{q+1} I$, et si $q = 0, B_1 = AB_0 + a_1 I$, avec la relation de définition qui impose : $B_0 = I$.

b **–** La démonstration est dans le cours. On a :

$$a_k = -\frac{1}{k}\left[S^k + a_1 S^{k-1} + a_2 S^{k-2} + \cdots + a_{k-1} S^1\right],$$
$$\text{avec} \quad S^m = \sum_{j=1}^{n} \lambda_j^m = \text{Trace}\,(A^m),$$

où les $\{\lambda_j\}$ sont les valeurs propres de A. Donc :

$$\begin{aligned} a_k &= -\frac{1}{k}\left\{\text{Trace}\,(A^k) + a_1 \text{Trace}\,(A^{k-1}) + a_2 \text{Trace}\,(A^{k-2}) + \cdots + a_{k-1} \text{Trace}\,(A)\right\} \\ &= -\frac{1}{k}\text{Trace}\,\left\{A^k + a_1 A^{k-1} + a_2 A^{k-2} + \cdots + a_{k-1} A\right\} \\ &= -\frac{1}{k}\text{Trace}\,\left\{A(A^{k-1} + a_1 A^{k-2} + a_2 A^{k-3} + \cdots + a_{k-1} I)\right\} = -\frac{1}{k}\text{Trace}\,\left\{AB^{k-1}\right\}. \end{aligned}$$

Remarque : $B_n = A^n + a_1 A^{n-1} + a_2 A^{n-2} + \cdots + a_n I = 0$ d'après le théorème de Cayley-Hamilton.

Le programme `souriau.c` exploite cette procédure.

```
> restart: dimension := 20:
> with(linalg):

  polcar:= proc(A,n,P)
  local i,j,u:
  global B;
  B:=evalm(A):
  for i from 1 to n
   do
   u := trace(B) /i:
   for j from 1 to n do B[j,j]:=B[j,j]-u :
   od:
   P[i]:= -u;
   B:=multiply(B,A);
  od:
  P[0]:=1
  end:
Warning, new definition for norm
Warning, new definition for trace
> Mat:= randmatrix(dimension, dimension);
```

$$Mat := \begin{bmatrix} -85 & -55 & -37 & -35 & 97 & 50 & 79 & 56 & 49 & 63 & 57 & -59 & 45 & -8 & -93 & 92 & 43 & -62 & 77 & 66 \\ 54 & -5 & 99 & -61 & -50 & -12 & -18 & 31 & -26 & -62 & 1 & -47 & -91 & -47 & -61 & 41 & -58 & -90 & 53 & -1 \\ 94 & 83 & -86 & 23 & -84 & 19 & -50 & 88 & -53 & 85 & 49 & 78 & 17 & 72 & -99 & -85 & -86 & 30 & 80 & 72 \\ 66 & -29 & -91 & -53 & -19 & -47 & 68 & -72 & -87 & 79 & 43 & -66 & -53 & -61 & -23 & -37 & 31 & -34 & -42 & 88 \\ -76 & -65 & 25 & 28 & -61 & -60 & 9 & 29 & -66 & -32 & 78 & 39 & 94 & 68 & -17 & -98 & -36 & 40 & 22 & 5 \\ -88 & -43 & -73 & 25 & 4 & -59 & 62 & -55 & 25 & 9 & 40 & 61 & 40 & -78 & 62 & 11 & 88 & 1 & 30 & 81 \\ -5 & -28 & 4 & -11 & 10 & 57 & -82 & -48 & -11 & 38 & -7 & 58 & -94 & -68 & 14 & -35 & -14 & -9 & -51 & -73 \\ -73 & -91 & 1 & 5 & -86 & 43 & -4 & -50 & 50 & 67 & -39 & 8 & -49 & 11 & 93 & -14 & -99 & -67 & 68 & 45 \\ 76 & 6 & 72 & -28 & -61 & -59 & 6 & -87 & 72 & -46 & -68 & -42 & -47 & -32 & 37 & -93 & -58 & -90 & -53 & -69 \\ -84 & 46 & 59 & -56 & -83 & -91 & 92 & -93 & 91 & -54 & 10 & -77 & -63 & -90 & 61 & -3 & -82 & 16 & -40 & 21 \\ -94 & -98 & 75 & 39 & 95 & -68 & 98 & -36 & -95 & 8 & 92 & 8 & -95 & -18 & 44 & 66 & -62 & 40 & 68 & -67 \\ 68 & -65 & 43 & 6 & 39 & -67 & 8 & 20 & 93 & 45 & 81 & 8 & -44 & -80 & -5 & 23 & 34 & -81 & -95 & 63 \\ -24 & -63 & -36 & 85 & 35 & -67 & 19 & 0 & 60 & 95 & 95 & -18 & 46 & -20 & 52 & 31 & 65 & 46 & -24 & 63 \\ 73 & 76 & -68 & 67 & -38 & 23 & 14 & 58 & -92 & -8 & 36 & -90 & 96 & -19 & -4 & 38 & -61 & -58 & -91 & 45 \\ -79 & -27 & 32 & -24 & -46 & 12 & 81 & 63 & -85 & -36 & -35 & 11 & 90 & 31 & 47 & -50 & -54 & 71 & -71 & -17 \\ 64 & -34 & 60 & 99 & 48 & 36 & 73 & -40 & 49 & 55 & 24 & 65 & 6 & -65 & 52 & 99 & -39 & 49 & -76 & -15 \\ 44 & -98 & 27 & 36 & 18 & 32 & 5 & 0 & 8 & 63 & -80 & -97 & 25 & -76 & -53 & -57 & -99 & 90 & 86 & -84 \\ 8 & -10 & 63 & -39 & -48 & -94 & 10 & -20 & 70 & 98 & 58 & 45 & -12 & 28 & 3 & 13 & -5 & -45 & -41 & 84 \\ -49 & -79 & 93 & -81 & -47 & 28 & 73 & -24 & 8 & 27 & -18 & 62 & 74 & 24 & 3 & 32 & 65 & -26 & 87 & 33 \\ 76 & -57 & -80 & -92 & 81 & 77 & -87 & 50 & 74 & -60 & 19 & -68 & 78 & 34 & 66 & -53 & 59 & 28 & 38 & 6 \end{bmatrix}$$

```
> depart:=time():
  polcar(Mat,dimension,P):
  fin := time() - depart;
  for i from 0 to dimension do if P[i]<>0 then print(P[i],x^(dimension-i)) fi od;
  fin := time() - depart;
```

$fin := 1.062$

$1, x^{20}$

$241, x^{19}$

$51185, x^{18}$

$5302845, x^{17}$

$982978574, x^{16}$

$484082521611, x^{15}$

$-162462470637262, x^{14}$

$-80812585835534825, x^{13}$

$-12548615159897043728, x^{12}$

$989317170031833105256, x^{11}$

$371517953645606749418465, x^{10}$

$-22346505217784284771587036, x^{9}$

$-5180627703327892785141875220, x^{8}$

$753307511188981105037618610502, x^{7}$

$189072668032386424809552431842706, x^{6}$

$-5948847609383046503201323860143824, x^{5}$

$-2255760916452562184138301417855106720, x^{4}$

$421803817826185215276969408891411138287, x^{3}$

$56457778805658467866663150570013977773491, x^{2}$

$-3875290559015621017122626962352576833313555, x$

$-140141767454063507288634400018311438395420160, 1$

$fin := 1.062$

```
> depart:=time():
  charpoly(Mat,x);
  fin := time() - depart;
```

$$-14014176745406350728863440001831143839542016 0 - 387529055901562101712262696235257683331355 5\,x + 564577788056584678666631505700139777349 1\,x^2 + 421803817826185215276969408891411138287\,x^3 - 22557609164525621841383014178551067 20\,x^4 - 594884760938304650320132386014382 4\,x^5 + 1890726680323864248095524318427 06\,x^6 + 75330751118898110503761861050 2\,x^7 - 518062770332789278514187522 0\,x^8 - 22346505217784284771587036\,x^9 + 3715179536456067494184 65\,x^{10} + 98931717003183310525 6\,x^{11} - 12548615159897043728\,x^{12} - 80812585835534825\,x^{13} - 162462470637262\,x^{14} + 484082521611\,x^{15} + 982978574\,x^{16} + 5302845\,x^{17} + 51185\,x^{18} + 241\,x^{19} + x^{20}$$

$$fin := 1.297$$

```
>
> evalm(B);     # donc B[n+1]
```

$$\begin{bmatrix}
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0\\
0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0&0
\end{bmatrix}$$

VI I) Résolutions d'équations

I) Des "solveurs" à notre disposition

```
> restart:eq1 := 2*x^2 + 3*x - 6: solve (eq1=0,x);evalf(");
```

$$-\frac{3}{4}+\frac{1}{4}\sqrt{57}, -\frac{3}{4}-\frac{1}{4}\sqrt{57}$$

$$1.137458609, -2.637458609$$

```
> eq2 := x^4 +4*x^3- 20*x^2 +4*x -21; solve (eq2=0,x);
```

$$eq2 := x^4 + 4\,x^3 - 20\,x^2 + 4\,x - 21$$

$$3, -7, I, -I$$

```
> sys:={3*x+5*y=8,x+y=-1};solve(",{x,y});x;assign("");x;     # l'assignation n'est pas automatique
```

$$sys := \{3\,x + 5\,y = 8, x + y = -1\}$$

$$\{y = \frac{11}{2}, x = \frac{-13}{2}\}$$

$$x$$

$$\frac{-13}{2}$$

```
> restart: plot({x,tan(x)},x=-1..12,y=-1..9,color=BLACK);
```



```
> plot(tan(x) - x, x= 4.3..4.7, y=-1..1);
```




```
> eq3:= tan(x) - x:solve(eq3,x);fsolve(eq3, x=6.5..7.9);
```

RootOf(tan(_Z) − _Z)

7.725251837

II) DICHOTOMIE "f(x) = 0

```
> restart: Digits := 32:
  f:=x->tan(x)-x:
  mini := evalf(Pi/2): maxi:= evalf(3*Pi/2):
  compteur:=1: eps:= 10^(-25):
  ecart:= abs(maxi  - mini ): :
  while ecart>eps
       do milieu:= (mini + maxi)/2:
          compteur:=compteur+1;
          if evalf (f(mini)* f(milieu)) >0
            then mini := milieu
            else maxi := milieu
          fi;
          ecart := abs(maxi - mini):
       od:
  solution:= milieu:
  printf(`Nombre d'itérations: %d    solution:   %.30f`,compteur,solution);
```

```
Nombre d'itérations: 86    solution:   4.493409457909064175307880876371
```

III) SECANTE "f(x) = 0" (itération de Lagrange)

```
> restart: Digits := 32:
> seco := proc(f,x0,x1,eps)
  local xa,xb, ya, yb ,x , dy, ecart, compteur;
  xa := x0: xb := x1 : compteur:=0:
  ecart := abs(xa - xb);
  while(ecart > eps)
  do
   compteur:=compteur+1:
   ya := evalf( f(xa)):
   yb := evalf( f(xb)):
   dy := yb- ya:
   if dy <> 0 then
     x:= xb - yb * (xb-xa) / dy;
     ecart := abs(xa - xb);
     xa := xb: xb := x:
    else ecart := 0
   fi:
  od:
  printf(`Nombre d'itérations: %d     solution:    %.30f`,compteur,xb)
  end:
> seco(x->tan(x)-x, 4.4, 4.5,10^(-25));
Nombre d'itérations: 9    solution:   4.493409457909064175307880927280
```

IV) NEWTON "f(x) = 0"

```
> restart:Digits:=30;
  newton := proc(f,x0,eps)
  local fprime, w1,w2, ecart, compteur;
  fprime:=D(f): w1 := evalf(x0): compteur:=0:ecart:= 1000:
  while(ecart > eps)
  do
   compteur:=compteur+1:
   w2 := evalf(w1 - f(w1) / fprime(w1));
   ecart := abs(w1 - w2);
   w1 := w2:
   print(w1)
  od;
  printf(`Nombre d'itérations: %d     solution:    %.30f`,compteur,w1)
  end:
```

$$Digits := 30$$

```
> newton(x->tan(x) - x,4.6,10^(-25));
```

4.54573212207901343402068282493

4.50614558803960722777136147291

4.49417163017409291058347335115

4.49341219682140713537425179900

4.49340945794444156672779209084

4.49340945790906417531378323036

4.49340945790906417530788092728

4.49340945790906417530788092728

```
Nombre d'itérations: 8    solution:   4.493409457909064175307880927280
```

V) POINT FIXE "f(x)=x f lipschitzienne"

```
> u[0]:=0.25:  f:=x->cos(x):
> for i from 1 to 25 do u[i]:=f(u[i-1]) od:
  'u[1]'=u[1];'u[5]'=u[5];'u[15]'=u[15];'u[25]'=u[25];
```

$$u_1 = .9689124217106447841 4$$

$$u_5 = .7872140049933128665 1$$

$$u_{15} = .7400235713587020831 5$$

$$u_{25} = .7391031897624653681 1$$

```
> F1:=plot({f(x),x},x=0..2):
> L:=[u[0],0,u[0],f(u[0])]: n:=25:
> for k from 2 to n do
  L:=[op(L),L[2*k]]:
  if type(k,odd) then L:=[op(L), f(L[2*k + 1])]
     else L:=[op(L), L[2*k+1]] fi od:
  l:=[]: for k from 0 to n do
  l:=[op(l),[L[2*k+1],L[2*k+2]]]:
  od:
> with(plottools):plots[display]({F1,CURVES(l)},view=[0..2,0..1]);
```



VI) ACCELERATION DE L'ITERATION: AITKEN-STEFFENSEN "f(x)=x f lipschitzienne"

Pour résoudre l'équation tan (x) = x on ne peut utiliser l' itération; en effet:

```
> D(tan)(x);
```

$$1+\tan(x)^2$$

est supérieur ou égal à 1.

On pose y = tan(x); pour x dans]Pi/2, 3Pi/2[, x = arctan(y) + Pi

On va résoudre arctan(y) + Pi = y

```
> g:=y->arctan(y) + Pi;
```

$$g := y \to \arctan(y)+\pi$$

```
> D(g);
```

$$y \to \frac{1}{1+y^2}$$

majoré par 1/(1+Pi²) < 1 dans [Pi, 3Pi/2]

```
> plot ({g(x),x, 3*Pi/2}, x=0..2*Pi, color=black);
```




```
> u[0]:= 3:
  for i from 1 to 50 do u[i]:=evalf(g(u[i-1]),40) od:
  'u[15]'=u[15];'u[25]'=u[25];'u[35]'=u[35];'u[50]'=u[50];
```

$$u_{15} = 4.493409457909064175279313375018425713920$$

$$u_{25} = 4.493409457909064175307880927280320517576$$

$$u_{35} = 4.493409457909064175307880927280322082215$$

$$u_{50} = 4.493409457909064175307880927280322082215$$

Accélération avec DELTA² d'AITKEN

rappel: si la suite (x[n]) converge vers 1, alors la suite (t[n]) converge plus rapidement vers 1 avec

t[n] = (x[n] x[n-2] - x²[n-1]) / (x[n-2] - 2 x[n-1] + x[n]) = x[n] - (x[n] - x[n-1])² / (x[n-2] - 2 x[n-1] + x[n])

```
> Digits:=40:
  u[0]:=3:u0:=u[0]:delta2 :=1:k:=0:
  while abs(delta2) > 10^(-40)
  do
   k:=k+1:
   u1:=evalf(g(u0)):
   u2:=evalf(g(u1)):
   delta := u1-u0:
   delta2 := u2 - 2*u1 + u0:
   if delta2<>0
    then
     u[k]:=evalf(u0- delta**2 / delta2):
     u0:=u[k]
    else u[k]:=u[k-1]
   fi:
  od:
  'k' = k;'u[k]'=u[k];
```

$$k = 5$$

$$u_k = 4.493409457909064175307880927280322082215$$

```
>
```

VIII) Calculs d'intégrales

I) Primitives:

```
> restart:Int(x/(x**2 + x + 1) , x)  = int(x/(x**2 + x + 1) , x) ;
  diff(rhs("),x); simplify(");
```

$$\int \frac{x}{x^2+x+1}\,dx = \frac{1}{2}\ln(x^2+x+1) - \frac{1}{3}\sqrt{3}\arctan\left(\frac{1}{3}(2x+1)\sqrt{3}\right)$$

$$\frac{1}{2}\frac{2x+1}{x^2+x+1} - \frac{2}{3}\frac{1}{1+\frac{1}{3}(2x+1)^2}$$

$$\frac{x}{x^2+x+1}$$

```
> I2:=int(x^3*exp(x-x^7),x); series(",x,8);  # pas d'expression simple
```

$$I2 := \int x^3\,e^{(x-x^7)}\,dx$$

$$\frac{1}{4}x^4 + \frac{1}{5}x^5 + \frac{1}{12}x^6 + \frac{1}{42}x^7 + \frac{1}{192}x^8 + O(x^9)$$

II) Intégrales définies:

```
> Int(1/sqrt(1-x^2),x = 0 .. 1)= int(1/sqrt(1-x^2),x = 0 .. 1);
```

$$\int_0^1 \frac{1}{\sqrt{1-x^2}}\,dx = \frac{1}{2}\pi$$

```
> int(x^3*exp(x-x^7),x=0..1);evalf(");
```

$$\int_0^1 x^3\,e^{(x-x^7)}\,dx$$

$$.3926483705$$

méfiance Maple peut se tromper!

```
> int (sqrt(sin(x)-x*cos(x))^2,x=Pi..2*Pi);  # il a remplacé sqrt(f^2) par f !!!
```

$$-4$$

```
> f:=u->sin(u) - u* cos(u): g:=u->f(u)*f(u): h:=u->sqrt(g(u)):
  int(h(x), x=Pi..2*Pi) ; evalf(");          # c'est mieux !!
```

$$\int_{\pi}^{2\pi} \sqrt{(\sin(x) - x\cos(x))^2}\,dx$$

$$9.641144954$$

III) Intégrales impropres:

```
> int(1/x^5, x=1..infinity);
```

$$\frac{1}{4}$$

IV) Intégrales doubles:

```
> Int(Int((x+y)*cos(x+y),x = 0 .. Pi/2),y = 0 .. Pi/2)=int(int((x+y)*cos(x+y),x = 0 .. Pi/2),y = 0 .. Pi/2);;
```

$$\int_0^{\frac{1}{2}\pi}\int_0^{\frac{1}{2}\pi} (x+y)\cos(x+y)\,dx\,dy = \pi - 4$$

V) Changement de variable:

```
> with (student):
> K:=Int(t/((t^2+1)*sqrt(1-t^4)), t=0..1): K = value(K);
```

$$\int_0^1 \frac{t}{(t^2+1)\sqrt{1-t^4}}\,dt = \frac{1}{2} + \left(\sum_{\_\alpha = \%1}\left(\frac{1}{8}\sqrt{2}\,\_\alpha\,\mathrm{EllipticPi}\left(\frac{1}{2}, \frac{1}{2}\sqrt{2}\right)\right)\right)$$

$$\%1 := \mathrm{RootOf}(\_Z^2 + 1)$$

un coup de main!

```
> changevar(t^2=u,K,u):
> K=value(");
```

$$\int_0^1 \frac{t}{(t^2+1)\sqrt{1-t^4}}\,dt = \frac{1}{2}$$

VI) Intégration par partie:

```
> L:=Int(x^2 * arccos(x),x);intparts(L,arccos(x))=value(");
```

$$L := \int x^2 \arccos(x)\,dx$$

$$\frac{1}{3}\arccos(x)\,x^3 - \int -\frac{1}{3}\frac{x^3}{\sqrt{1-x^2}}\,dx = \frac{1}{3}\arccos(x)\,x^3 - \frac{1}{9}x^2\sqrt{1-x^2} - \frac{2}{9}\sqrt{1-x^2}$$

VII) Sommes de RIEMANN

```
> leftbox(cos(x), x=0..Pi/2,10);
```




```
> middlesum(cos(x) , x=0..Pi/2,10);
```

$$\frac{1}{20}\pi\left(\sum_{i=0}^{9}\cos\left(\frac{1}{20}\left(i+\frac{1}{2}\right)\pi\right)\right)$$

```
> Digits:= 40:value(""); evalf(");
```

$$\frac{1}{20}\left(\cos\left(\frac{1}{40}\pi\right)+\cos\left(\frac{3}{40}\pi\right)+\frac{1}{2}\sqrt{2+\sqrt{2}}+\cos\left(\frac{7}{40}\pi\right)+\cos\left(\frac{9}{40}\pi\right)+\cos\left(\frac{11}{40}\pi\right)+\cos\left(\frac{13}{40}\pi\right)+\frac{1}{2}\sqrt{2-\sqrt{2}}+\cos\left(\frac{17}{40}\pi\right)+\cos\left(\frac{19}{40}\pi\right)\right)\pi$$

$$1.001028824142708521899310360626328219651$$

VIII) PROGRAMMATION

a) Rectangles (point médian)

```
> median := proc(f,a,b,n)
  local eps,x,s,i:
  eps := (b-a)/n: # le pas de la subdivision
  x:= a + eps/2:   # milieu du premier intervalle
  s:=0:
  for i from 1 to n
  do
   s:=s + f(x):
   x := x + eps: # on passe à l'intervalle suivant
  od:
  RETURN(s*eps):
  end:
> median (cos, 0,Pi/2,10);evalf(");median (cos, 0,Pi/2,1000):evalf(");
```

$$\frac{1}{20}\left(\cos\left(\frac{1}{40}\pi\right)+\cos\left(\frac{3}{40}\pi\right)+\frac{1}{2}\sqrt{2+\sqrt{2}}+\cos\left(\frac{7}{40}\pi\right)+\cos\left(\frac{9}{40}\pi\right)+\cos\left(\frac{11}{40}\pi\right)+\cos\left(\frac{13}{40}\pi\right)+\frac{1}{2}\sqrt{2-\sqrt{2}}+\cos\left(\frac{17}{40}\pi\right)+\cos\left(\frac{19}{40}\pi\right)\right)\pi$$

$$1.001028824142708521899310360626328219651$$

$$1.000000102808386576708613952908879177914$$

b) Trapèzes

```
> trapeze := proc(f,a,b,n)
  local eps,x,s,i:
  eps := (b-a)/n: # le pas de la subdivision
  s:= f(a) /2:
  x:=a+ eps:
  for i from 1 to n-1
  do
   s:=s + f(x):
   x := x + eps: # on passe à l'intervalle suivant
  od:
  s:=s + f(x)/2:
  RETURN(s*eps):
  end:
> trapeze(cos,0,Pi/2,10):evalf(");trapeze(cos,0,Pi/2,1000):evalf(");
```

.9979429863543572270076132958726044148990

.9999997943832331883209353231083563450935

c) Simpson

```
> simps := proc(f,a,b,n)
  local eps,x,s,i:
  if type (n,odd) then ERROR (`nombre pair de subdivisions!`) fi;
  eps := (b-a)/n: # le pas de la subdivision
  s:= f(a):
  x:=a+ eps:
  for i from 1 to n-1
  do
    if type (i,odd) then s:=s + 4* f(x) else  s:=s + 2* f(x) fi:
    x := x + eps: # on passe à l'intervalle suivant
  od:
  s:=s + f(b):
  RETURN(s*eps/3):
  end:
> simps(cos,0,Pi/2,10):evalf(");
```

1.000003392220900552095136890858273760393